
Chapter 1

Finding the information you need
This book, Programming in OpenScript, explains how to use the
OpenScript programming language. An overview of the topics is
available in this chapter, as well as an explanation of the other re-
sources available for learning about OpenScript.

Looking at what’s inside . . . . . . . . . . . . . . . . . . . . . . . . . . . 10
Finding other resources . . . . . . . . . . . . . . . . . . . . . . . . . . . . 13

Looking at what’s inside
This online guide contains a wealth of information on OpenScript,
the programming language that you can use to enhance the
applications you create with ToolBook Instructor.™ This guide
provides a basic foundation for understanding OpenScript, and then
builds on that foundation, chapter by chapter. In addition to this
chapter, here is what you will find:

◆ Chapter 2, “Understanding OpenScript programming,” introduces
OpenScript and provides an overview of the ToolBook scripting
tools.

◆ Chapter 3, “Learning OpenScript basics,” explains the fundamen-
tal concepts, vocabulary, and techniques you need to know to
create scripts.

◆ Chapter 4, “Using the scripting tools,” describes how to use the
ToolBook scripting tools.

◆ Chapter 5, “Handling messages, events, objects, and properties,”
teaches you how to work with built-in messages, system objects,
and system properties, and how to create your own messages,
user properties, and self-contained objects.

◆ Chapter 6, “Creating an OpenScript statement,” describes the
components that make up a statement and the control structures
you use to control blocks of statements.

◆ Chapter 7, “Using variables, values, and text,” focuses on how
to use individual components of OpenScript programs, such as
variables, constants, and literal values, and how to use text in
your scripts.

◆ Chapter 8, “Writing common scripts,” teaches you how to use
OpenScript to accomplish many common tasks, such as checking
for errors.

◆ Chapter 9, “Handling user events,” discusses how to work with
events that a user generates by clicking a mouse button or typing
at the keyboard.

◆ Chapter 10, “Managing data,” provides information on using
OpenScript to perform several data-management tasks, such as
validating user input.

◆ Chapter 11, “Using Special effects,” describes how to use
OpenScript to include special effects, such as animation, in your
application.

◆ Chapter 12, “Using dynamic data exchange,” explains how
to send and receive data from other Microsoft® Windows®

applications.

◆ Chapter 13, “Using dynamic-link libraries,” shows how to extend
ToolBook and OpenScript by calling Windows functions and other
utility programs, and by translating Windows messages into
ToolBook messages.

◆ Chapter 14, “Debugging OpenScript programs,” presents tools
and techniques for building more reliable applications.



12

Identifying visual cues

To help you interpret the information it contains, this guide uses the
visual cues listed in the table below.

Visual cues

Example format Meaning

CTRL+ALT+DELETE Keys separated by a plus sign indicate keys to
press in combination.

comboBox In a paragraph, italic type indicates OpenScript
keywords.

button id 12 This font indicates an OpenScript code example.

\ A backslash indicates that an OpenScript
statement continues on the next line.

➥ This symbol indicates that OpenScript syntax is
continued from the previous line.

<isShift> In a code example or syntax statement, angle
brackets indicate a parameter that represents a
literal value or expression.

--Initialize In a code example, two hyphens precede
a comment.

before|after In a syntax statement, vertical bars separate
the options from which you can choose.

end [step] In a syntax statement, square brackets indicate
optional words and parameters.

… In OpenScript syntax, an ellipsis indicates
that the parameter can be repeated. In code
examples, ellipses indicate the presence of
any number of statements.

? This icon is located by text that indicates where
to look in online Help for more information.

Finding other resources
In addition to this guide, there are a variety of other resources
available for learning about OpenScript and ToolBook Instructor.
These are:

◆ the ToolBook web site, which contains a Knowledge Base, an FAQ
list, information on sample files, patches, and newsgroups. To visit
the ToolBook web site, go to http://www.toolbook.com/

◆ the ToolBook online Help system, which includes an OpenScript
reference of objects, properties, keywords, and dynamic-link li-
brary (DLL) functions, and also a comprehensive glossary. To use
online Help, press F1 from anywhere in ToolBook, or select an
item from the ToolBook Help menu.

Chapter 2

Understanding OpenScript programming
This chapter introduces OpenScript, the ToolBook programming
language, and describes how to use it to develop applications. This
chapter also includes an overview of the tools you can use to create
OpenScript programs, and a comparison of OpenScript and
other popular programming languages.

About OpenScript . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 16
OpenScript development tools . . . . . . . . . . . . . . . . . . . . . . 17
Comparing OpenScript with other languages . . . . . . . . . . . 20
Extending ToolBook and OpenScript . . . . . . . . . . . . . . . . . 21

About OpenScript
OpenScript is the ToolBook programming language. When you
create applications in ToolBook, you use ToolBook’s built-in tools
for tasks such as drawing objects on pages and creating hyperlinks.
To take full advantage of ToolBook capabilities, you can use
OpenScript to control the behavior of your application more directly
and precisely.

OpenScript is a full-featured programming language that includes
commands to accomplish a wide variety of tasks, from creating and
managing new objects to linking functions in Windows dynamic-link
libraries (DLLs). Though a powerful language, OpenScript is also easy
to use because of its syntax (similar to English), its wide range of
commands, and its object-oriented nature. Using ToolBook and
OpenScript, you can program sophisticated Windows applications in
a fraction of the time and effort needed to create similar applications
in C or C++.

OpenScript is fully integrated into the ToolBook environment.
You write scripts in the ToolBook script editor or enter commands di-
rectly into the Command window. The programs are compiled
using the ToolBook compiler, and ToolBook runs the finished scripts
at Reader level as part of your application. Because OpenScript is an
integral part of ToolBook, the processes of creating, testing, and
maintaining your programs are greatly simplified compared with the
same processes in other programming languages.



17

OpenScript development tools
You enter OpenScript statements in the script editor or the
Command window to create a script. One quick way to build a
script is to copy an existing object that has the script you want
(such as an object in a sample application), and then paste the
object into your application.

Working with the features of the Debugger, you can set
breakpoints and and trace the execution of statements in any
script. On the following pages, you’ll find a description of the
Debugger as well as the other scripting tools.

.

◆ Script editor Use to type new scripts and handlers, edit exist-
ing scripts, or cut, copy, and paste script segments from the
Clipboard. Editing a script in the script editor is much like edit-
ing a document in a simple word processing program. In addi-
tion, you can check a script’s syntax, print the script, and
display the ToolBook Debugger window.

◆ Command window Use to send messages, display information,
and test short script segments. You can enter one or more
OpenScript statements in the Command window and execute
them immediately.

◆ ToolBook Debugger window Use to correct errors in your
scripts. You can set and clear breakpoints, display the values of
local and system variables, and trace execution of script state-
ments and calls to other handlers.

For details about using the script editor and Command window,
see Chapter 4, “Using the scripting tools” For details about the
ToolBook Debugger, see Chapter 14, “Debugging OpenScript pro-
grams.”

Comparing OpenScript with other languages
If you have used another programming language, you will find
that OpenScript provides virtually all the features you are familiar
with from languages such as C, BASIC, and Pascal.

OpenScript C BASIC Pascal

if/then/else if/else If/Then/Else If/Then/Else

step for For/Next For/Do

while while While/Wend While/Do

do/until do/while Do/Loop Until Repeat/Until

conditions/when switch/case Select Case/Case Case/Of

increment += inc

decrement –= dec

send <message> void function Sub procedure

to get <message> function Function function

system books libraries libraries (or DLLs) units

system variables global variables global variables global
variables

break break, return

format (number) printf print using

The following list tells you where to look in this manual for more
information on several important ToolBook features.

◆ Scope of variables Variables can be local or global in
OpenScript. For details, see Chapter 7, “Using variables, values,
and text.”

◆ Data typing Variables are not typed by default, though you can
assign data types for efficiency. If variables are not typed, you can
use the data they contain alternately as strings and numbers. For
details, see Chapter 7, “Using variables, values, and text.”

◆ Arrays You can define one-, two-, or many-dimensional arrays,
or arrays containing variable numbers of elements. For details,
see Chapter 7, “Using variables, values, and text.”

◆ Data formatting You can format numbers, dates, and times
to alter how they are displayed and to make it easier to perform
calculations with them. For details, see Chapter 10, “Managing
data.”

◆ User properties Each object you create has built-in properties,
and you can also define user properties to hold values that you
create. For details, see Chapter 5, “Handling messages, events,
objects, and properties.”

◆ System books You can establish any ToolBook file as a system
book, or library, making its functions and handlers available to
any ToolBook application. For details, see Chapter 8, “Writing
common scripts.”

◆ User-defined functions In addition to using the built-in functions
of ToolBook, you can define your own functions to calculate
values, perform operations, or carry out other tasks. For details,
see Chapter 6, “Creating an OpenScript statement.”

◆ Notify handlers Using the notifyAfter and notifyBefore message
handlers, you can create objects that you can put into any
ToolBook application without altering any scripts. For details, see
Chapter 5, “Handling messages, events, objects, and
properties.”

◆ Dynamic script creation Using the execute statement and
evaluate function, you can create and execute scripts from within
other scripts while your application is running. For details, see
Chapter 8, “Writing common scripts.”

Extending ToolBook and OpenScript
You can extend OpenScript by interacting with other Windows appli-
cations, and even mainframe computers, by using:

◆ the Windows dynamic data exchange (DDE) protocol to incorpo-
rate the strengths of other Windows applications into your
ToolBook application. For example, an application might get the
results of complex calculations from Microsoft Excel and display
the results graphically in ToolBook. For details, see Chapter 12,
“Using dynamic data exchange.”

◆ the linkDLL command to link a Windows DLL and call its
functions. ToolBook can use functions from 32-bit Windows DLLs;
you need to know what functions are in the DLL and the data
types of their parameters, but after that you can use them like
any ToolBook function. For details, see Chapter 13, “Using dy-
namic-link libraries.”

◆ the translateWindow32 command to translate Windows messages
into ToolBook messages so that your ToolBook application can
respond to events generated by the Windows system. For details,
see Chapter 13, “Using dynamic-link libraries.”



Chapter 3

Learning OpenScript basics
This chapter is a primer on OpenScript that explains the fundamental
concepts, vocabulary, and techniques used to create scripts, and
presents simple examples of all the OpenScript elements discussed.

The chapter assumes that you already know how to use the
ToolBook user interface to create objects and change their proper-
ties, and that you have a basic understanding of general program-
ming concepts, including statements, variables, functions, and
control structures (such as if/then/else). You can find details about
how each of these elements is implemented in OpenScript later in
this book.

About scripting in ToolBook . . . . . . . . . . . . . . . . . . . . . . . . 24
Writing scripts . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 24
The object hierarchy . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 28
Objects and properties . . . . . . . . . . . . . . . . . . . . . . . . . . . . 33
Types of handlers . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 38
Entering and displaying data . . . . . . . . . . . . . . . . . . . . . . . . 39

About scripting in ToolBook
ToolBook is an object-oriented programming environment, which
means that you build your application by creating objects using the
ToolBook user interface. For example, buttons, rectangles, and fields
are all objects. They reside on pages and backgrounds, which are also
objects, and all of these together reside in a book, which is an object
as well. You set an object’s properties to define its appearance (color,
shape, text) and behavior (for example, what happens when the user
clicks it).

Like Windows, ToolBook is event-driven. ToolBook waits for an event
to occur, such as when a user presses a key, clicks the mouse, opens
a ToolBook file, chooses a menu item, moves the cursor across an
object, or even does nothing (an idle event).

Each event creates one or more messages in ToolBook. ToolBook
sends a message to the target, which is the object directly affected
by the event. For example, when the user goes to a new page in
the book, the enterPage message is sent to the new page, and the
leavePage message is sent to the original page. When the user clicks
an object, ToolBook sends the buttonClick message to the object
indicating that the user pressed and released the left mouse button.
If the user is doing nothing, ToolBook sends the idle message con-
tinuously to the current page.

Writing scripts
ToolBook has built-in responses to many of the messages generated
by events. For example, its response to the message sent when a user
chooses a built-in menu command is to display a dialog box
or perform an action.

For many events, you can rely on the built-in responses of
ToolBook. However, if you want to control directly how an object
responds to an event, you create a handler for a message. A handler
is a series of OpenScript statements that define the response to a
particular message. For example, this handler defines how an object
will respond when a user clicks it:

to handle buttonClick
request “I’ve been clicked.”

end buttonClick

Handlers are part of the script property of an object. A script can
contain any number of handlers, each responding to a different
message.

Any object in a ToolBook application, including books, pages, and
backgrounds, can have a script. To create a script, you use the
script editor, which you invoke much the same way you do the
dialog box for other properties of an object.

To create a simple handler:To create a simple handler:To create a simple handler:To create a simple handler:To create a simple handler:

1 In a new ToolBook book, add a button to a page.

2 To display the script editor, right-click the button, and then click
the Script button on the shortcut menu toolbar.

3 Enter this script:

to handle buttonClick
request "I’ve been clicked."

end buttonClick

4 To save the script, choose Update Script & Exit from the File
menu, or press CTRL+S.

When you save a script, ToolBook automatically compiles it. If
there is an error in the script, ToolBook cannot compile it and dis-
plays an error message. You cannot run a script until it has com-
piled successfully.

5 Switch to Reader level (press F3), and then click the button.

ToolBook displays a message box with the text “I’ve been
clicked.”

The handler you wrote above is for the buttonClick message, which
is sent when you click the left mouse button. Without the script,
the button does nothing when you click it; by adding this handler
to the button’s script, you define the button’s response to the click.

The entire script for an object is limited to that which can be
displayed in the script editor (64 KB). However, handlers can be
converted to simple wrappers that can call more detailed scripts
stored in other objects. For example:

to handle buttonClick
send MyHandleButtonClick to page 2

end

In this script, the detailed script to handle the buttonClick message
is handled in the MyHandleButtonClick handler on page 2.

Using the Command window to enter scripts

You can also run OpenScript statements at Author level using the
Command window, shown below. You can type one or more
OpenScript statements in the Command window to make an imme-
diate change to your application or to see the effect of an
OpenScript command. Use the Command window to quickly set
properties for objects or to test statements as you build your appli-
cation.

 Figure 1  The Command window

You can open the Command window by:

◆ choosing Command from the View menu.

◆ pressing SHIFT+F3.

◆ clicking the Command Window button on the toolbar.

For details about using the script editor and the Command window,
see Chapter 4, “Using the scripting tools.”

The object hierarchy
When an event occurs, ToolBook sends the corresponding message
to the target object. If the target object’s script has a handler for the
message, ToolBook runs the handler and the message travels no
further (unless the handler forwards the message, as explained later
in this chapter). If the object does not have a handler for the message,
the message passes to another object. The order in which a message
goes from object to object is called the object hierarchy.

In the object hierarchy, a message passes from an object to its parent
object. A page is the parent for objects and groups that reside on it,
the background is the parent for the page, and the book is the parent
for the background. If the object has no script for the message, the
message moves further up the hierarchy until it either encounters
the handler it is seeking or reaches the ToolBook system at the top
of the hierarchy. If the message reaches the system level, ToolBook
provides the default behavior (which might be to do nothing).

Messages enter the object hierarchy at different levels, depending
on the message. For example, most mouse-event messages are
sent to the object under the cursor. Menu-event messages enter the
hierarchy at the page level, as do navigation messages such as
enterPage, leavePage, next, and previous.

Note When a message enters the hierarchy at the page level, an
object with a notifyBefore handler for that message will intercept
the message before it starts at the page level. It’s important to keep
this in mind when writing self-contained objects. For more informa-
tion on creating self-contained objects using notify handlers, see
Chapter 5, “Handling messages, events, objects, and properties.”

The following procedure demonstrates how the object hierarchy
works. Before starting, create the button and script discussed in
“Writing scripts” earlier in this chapter.

To show the object hierarchy:To show the object hierarchy:To show the object hierarchy:To show the object hierarchy:To show the object hierarchy:

1 Add more objects to the same page: another button, a rectangle,
and a circle.

2 Make sure no object is selected, and then click the Script button
on the toolbar.

ToolBook displays the script editor with the script for the page,
which is the default when no objects are selected.

3 Enter this handler in the page’s script:

to handle buttonClick
request "Your click reached the page."

end buttonClick

4 To save the script, choose Update Script & Exit from the File
menu, or press CTRL+S.

5 Switch to Reader level (press F3). Then, to see the results, click
on each object and on a part of the page that contains no objects.

There are now two buttonClick handlers in your book: the one you
created earlier for the button and the one you just entered as part
of the page script. If you click the button with its own handler,
ToolBook runs the handler for that button. However, if you click an
object without a handler for the mouse event, the message passes
up the hierarchy to the page, where ToolBook runs the handler you
just entered.

Forwarding a message

If an object has a handler for a message, the message stops at that
object. However, you can forward a message to send it further up
the object hierarchy so that the message triggers additional handlers
or the ToolBook default response.

For example, if a button on the page handles a buttonClick message,
the message stops with the button. If there is another handler for
the buttonClick message higher in the hierarchy, the second handler
will not run because it will never receive the message. If you want
both handlers to run, use the forward command in a handler to pass
the message up the hierarchy.

Use the script editor to modify the script you created earlier for the
button by adding a forward command:

to handle buttonClick
request "I’ve been clicked."
forward

end buttonClick

When you click the button at Reader level, you will get two mes-
sages: one produced by the handler you just modified and one
produced by the handler you entered earlier for the page.

Forwarding a message is often necessary when ToolBook already
has a default response to a message, as for keystrokes. For example,
you can capture keystrokes with a handler for the keyChar event in
a page script. When you do, however, ToolBook does not get the
keystroke messages, and so it will not display the keystrokes unless
you also forward the messages.

For example, create a field and ensure Enabled is selected on its
right-click shortcut menu. Then, add this handler to the script of
the field:

to handle keyChar
beep 1
forward

end

When you enter data into the field at Reader level, each keystroke
causes a beep. In addition, the messages caused by the keystrokes
pass up the object hierarchy to the ToolBook system. The ToolBook
default response is to display the correct characters in the field. If
you remove the forward statement, your handler runs, but ToolBook
does not receive the messages and no characters are displayed.

Using the object hierarchy effectively

You can take advantage of the object hierarchy to make writing
and debugging scripts easier. For example, you can use the forward
command to pass messages up the object hierarchy. This allows
multiple handlers to respond to a single message. You can also put
a message handler in a page or book script so that no matter what
the target object is, you can handle the message with a single script.
Finally, you can selectively disable events to prevent handlers
at a higher level from taking effect by not including the forward com-
mand.

Objects and properties
Everything about an object, including its position, size, color, and
script, is defined by a property. You can view and set many object
properties using the ToolBook user interface. For example, you can
set the color of an object by clicking a tile in the color tray. How-
ever, you often get and set properties using OpenScript statements.

For example, you can set the color of an object to red as a warning,
or get the text of a field after the user enters data. This section
explains how you refer to objects and properties using OpenScript
statements.

? For a complete list of ToolBook objects and their properties, refer to
the OpenScript reference in online Help.

NoteNoteNoteNoteNote ToolBook has system objects with their own properties, such
as the palettes. For details, see Chapter 5, “Handling messages,
events, objects, and properties.”

Referring to objects in scripts

If you want to set or get the properties of an object in a script, you
must identify the object by type and name. To identify the type, use
OpenScript keywords such as button, page, and roundedRectangle.
For example:

caption of button ID 3
objectCount of page "Contents"
bounds of roundedRectangle ID 1

To refer to a particular object, assign a name to the object in its
Properties dialog box, and then use the name as follows:

to handle leavePage
hide field "User Tip"
go to page "Contents"
forward

end

You must use quotation marks around an object name if it includes
spaces or special characters other than an underscore. For example,
you must use quotation marks around the name "Planting Schedule"
but not around Garden_Plan. However, you should use quotation

marks around names in all cases because ToolBook runs faster if
it does not have to determine whether the object name is a variable.

You can also refer to an object by its ID number, which is assigned
by ToolBook when you create the object, as in this example:

button ID 1 of page ID 0

When practical, however, you should use a name to refer to an
object. If you copy and paste an object, its ID number changes
because ToolBook creates a new version of it, but its name
remains the same.

Using implicit and explicit references

If the object you are referring to is on the current page (the page
displayed in the current viewer), you can use an implicit reference
to your object. An implicit reference identifies only the object type
and name. This makes your scripts short and easy to read. However,
if the object is on the background, on another page, or in another
book, you need to use an explicit reference that includes the location
of the object. The following statements illustrate explicit references:

--On the background
bounds of recordField "Name" of background ID 1

--On another page
text of field "Name" of page "Introduction"

--In another book
script of group "clock" of page 3 of book "widgets.tbk"

To determine the correct explicit reference for an object, move
the cursor over it at Author level. ToolBook displays the explicit ref-
erence of the object in the status bar.

Every object has a uniqueName property that contains its explicit

reference. The uniqueName is a formal object reference; when
OpenScript commands and functions return the name of an object,
the name is always the uniqueName property of the object.

Referring to the current object

You can refer to the current object using the special terms this, my,
self, target, and selection:

◆ this refers to the page or background displayed in the current
viewer, or to the current book. For example:

bounds of recordField "Name" of this background

pageCount of this book

◆ my and self in a handler refer to the object whose script contains
that handler. For example:

request my uniqueName

move self by 1000,1000

◆ target refers to the object that first received the message that
triggered the current handler. For example, if the user clicks a
pushbutton, that button is the target, even if the message is for-
warded up the object hierarchy to the book. You can use target
to write scripts that handle the same event for many different
objects, as in this handler for a book script:

--Determines what type of object the user clicked
to handle buttonClick

if target contains "button" then
request "You clicked a button."

end if
if target contains "field" then

request "You clicked a field."
end if

end buttonClick

◆ selection refers to the currently selected object or objects.
For example:

move selection by 2880,-2880

Setting and getting the values of properties

You assign a value to a property of an object using the assignment
operator (=) or the command set or put. All three methods accomplish
the same task, so you can interchange them as you like. The follow-
ing examples illustrate how to use all three:

fillColor of ellipse "face" = red

set fillColor of ellipse "face" to red

put red into fillColor of ellipse "face"

Many properties have constraints on the values you can assign to
them. For example, the property transparent can only be set to the
values true or false. If you attempt to assign an invalid value, an er-
ror results.

Except for ID numbers, the properties of objects are persistent. This
means that the properties of an object are stored with and accompany
the object when it is cut or copied. Some properties are read only;
for example, a book’s pageCount property. You can get the value
of the property, but ToolBook does not allow you to alter it.

Note If you misspell the name of a property when you assign a value
to it, ToolBook creates a user property with that name instead of re-
porting an error. For details, see Chapter 5, “Handling messages,
events, objects, and properties.”

To get the value of a property, use it in any expression:

if fillColor of ellipse "face" is red then
fillColor of ellipse "face" = blue

end

You can also use a get command to put the value of a property into
a special variable called It:

get fillColor of ellipse "face" --Puts the value of fillColor
-- into the variable It

For details about the variable It, see Chapter 7, “Using variables,
values, and text.”

Types of handlers
The most common type of handler in ToolBook responds to mes-
sages and begins with to handle. However, there are actually five
types of handlers in ToolBook:

◆ to handle responds to a message.

◆ to get returns a value for use in other handlers. Use this type
of handler to create a user-defined function or to get the value
of  a property defined using OpenScript statements. For details,
see Chapter 5, “Handling messages, events, objects, and properties”
and Chapter 6, “Creating an OpenScript statement.”

◆ to set defines new properties with OpenScript statements. For
details, see Chapter 5, “Handling messages, events, objects, and
properties.”

◆ notifyBefore or notifyAfter defines a handler that is notified
when a page receives a particular message. Use these handlers
to create objects that can be copied into an application without
having to change scripts anywhere in the book. For details, see
Chapter 5, “Handling messages, events, objects, and properties.”

Using parameters with handlers

A parameter is information that accompanies a message to provide
more detail. For example, ToolBook sends three parameters with
a buttonClick message, two of which indicate whether the user was
holding down the SHIFT or CTRL key when the button was clicked.
You can include variables on the opening line of the handler to re-
ceive parameters:

--Displays a message if the user clicks while pressing SHIFT;
--the variable's location, isShift, and isCtrl are parameters
to handle buttonClick location, isShift, isCtrl

if isShift is true then
request "You clicked while the Shift key was down."

end if
end buttonClick

Entering and displaying data
The user generally interacts with your application through the
objects you create—clicking buttons, entering text into fields, or
choosing menu items. At times, however, you might need a more
direct way to get data from the user or to display text. For example,
you might need to display the values of variables, display an error
message, or prompt the user for a password. You can use the:

◆ Command window to enter OpenScript statements or display
values.

◆ ask command to solicit text from the user.

◆ request command to display a message and prompt the user
to click a button in response.

◆ text of fields and record fields to display text or the values
of variables.

◆ status bar, in which you can display any text.

◆ dialog boxes that you create to prompt the user for information.

Using the Command window to display information

You can execute OpenScript statements directly in the Command
window; ToolBook displays the results immediately. You can also
use the Command window to view data. If you use the put statement
without naming a container to put the data into, ToolBook displays
the data in the Command window:

put pageCount of this book --Shows page count in
    --the Command window

Prompting for information with the ask command

To prompt the user to enter a single line of data, use the ask com-
mand, which displays a dialog box with OK and Cancel buttons. The
user’s response is returned in the variable It:

to handle enterBook
ask "Enter your name:"
if It is not null then

request "Welcome to the Tutorial," && It
end if
forward

end enterBook

Asking a yes or no question with the request command

If you just want a quick yes-or-no answer from the user, use the
request command to display a message and buttons for Yes and No.
The text of the button the user clicks is returned in the variable It:

request "Do you wish to move to the next page?" with "Yes" or "No"
if It is "Yes" then

send next
end

Setting the text of fields and record fields

If you want to display information as part of the current page, set
the text of a field or record field. The text is a property of the field
or record field, so it retains its value until you change it again. For
example, this handler shows the name of the current page in a
record field, updating it as the user navigates from page to page:

to handle enterPage
text of recordField "Caption" = name of this page
forward

end

Setting the caption of the status bar

To display text in the status bar at the bottom of the ToolBook main
window, set its caption property. Using the status bar to display infor-
mation is especially effective when you do not want to interrupt the
flow of the application with a message box. To display the status bar
at Reader level, press F12, or use the command show statusBar in a
script or in the Command window. For example:

--Identifies the clicked object in the status bar
to handle buttonClick

caption of statusBar = "Current object name is" && target
end

Creating dialog boxes

The most flexible way to collect information from a user is to
display a dialog box. The dialog box can contain multiple fields,
buttons, combo boxes, and other controls, and it can display
default data and provide data validation of user entries. You can
create viewers that act like dialog boxes.

? For details about viewers, refer to the OpenScript reference and
other viewer topics in online Help.

Chapter 4

Using the scripting tools
To make your scripting and application development easier,
ToolBook includes a script editor and the Command window. This
chapter describes how to use these scripting tools and provides tips
for preventing common mistakes.

Using the script editor . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 44
Working in the Command window . . . . . . . . . . . . . . . . . . . 54
Preventing common scripting errors . . . . . . . . . . . . . . . . . . 59

Using the script editor
You can use the script editor to write, edit, check syntax for, and
save the script of a selected object. Like the ToolBook main
window, the script editor has a toolbar with buttons for the most
frequently used commands for writing and editing scripts.

Figure 1  The Script editor

Buttons on the script editor toolbar

Default script editor Script editor button when
button CTRL key is pressed

Save

Check Syntax

Debug

Undo

Redo

Cut

Copy

Paste

Comment Uncomment

Indent Outdent

Find

Find Next

Replace

Parent

You open the script editor at Author level by:

◆ selecting an object and then clicking the Script button on the toolbar.

◆ right-clicking an object and then clicking the Script button on the
shortcut menu toolbar.

◆ selecting an object, choosing the appropriate Properties command
from the Object menu, and then clicking the Script button on the
script editor toolbar.

◆ typing edit script of <object> in the Command window.

◆ pressing CTRL and then double-clicking an object.

◆ in the active script editor, clicking the Parent button and choosing
an object from the pop-up menu.

Tip To create a page script, press CTRL and click the status bar. To
create a background script, press CTRL+ALT and click the status bar.
To create a book script, press ALT and click the status bar.

Editing multiple scripts

You can open one script editor for each object you want to work
with. Script editors are modeless, except when associated with
shared scripts. Modeless script editors allow you to perform other
ToolBook tasks while they are open, and jump from one open script
editor to another to work on the contents of several scripts simulta-
neously. To move between modeless script editors, click the one you
want to work in, or choose one from the current script
editor Window menu.

Tip You can display the script editor and modify a script at Reader
level in two ways. You can execute the OpenScript statement edit
script of <object> from the script of another object or from the Com-
mand window. Or, when Show right-click menus at Reader level is
selected in the Options dialog box, you can right-click the object and
click the Script button on the object’s shortcut menu toolbar.

Note The ability to edit scripts at Reader level is disabled in the
run-time version of ToolBook.

Writing scripts

You can write a script for any ToolBook object by opening the
object’s script editor and entering the script as you would with any
text editor. Use these guidelines when you write a script:

◆ Lines can be any length, but lines over 80 characters are truncated
when you print the script. Split a statement across more than one
line by using the backslash (\) as a continuation character between
expressions. (You cannot use a backslash in the middle of a
string.) For example:

ask "Enter your complete address."

◆ Put multiple statements on one line by separating statements
with semicolons (;).

◆ Indent handlers to make the script easier to read.

◆ If the previous line was indented, the script editor automatically
indents to the same point when you press ENTER. Press BACK-
SPACE to remove an indent.

◆ Insert comments in the script by typing two hyphens (--) before
the comment text, or use block comments on selected text.

You can also write scripts by putting script text into the script
property of an object. For details, see Chapter 8, “Writing common
scripts.”

Using script editor commands

The following tables list the menu commands, toolbar buttons, and
keystroke shortcuts you use in the script editor while writing scripts.

Use the following commands to move the insertion point.

Script editor commands for moving the insertion point

To move Press

Any direction Arrow keys

Left one word CTRL+LEFT ARROW

Right one word CTRL+RIGHT ARROW

Between the beginning HOME
of a line and the first
nonspace character of a line

To the end of a line END

To the previous screen PAGE UP

To the next screen PAGE DOWN

To the beginning of the script CTRL+HOME

To the end of the script CTRL+END

To the top of the current screen CTRL+PAGE UP

To the bottom of the current screen CTRL+PAGE DOWN

Use the following commands to select text.

Script editor commands for selecting text

To Key command Menu command

Select to the start SHIFT+HOME
of a line

Select to the start SHIFT+CTRL+HOME
of the script

Select a single word Double-click

Select a block of text Click and drag

Select the contents SHIFT+F9 Select All from the
of the script editor Edit menu

Extend a selection SHIFT+arrow keys

Cancel a text selection Click elsewhere

Use the following commands to work with text and scripts.

Script editor commands for working with text

Key Toolbar Menu
To command button command

Undo changes CTRL+Z Undo from
Edit menu

Reverse an Undo CTRL+A Redo from
Edit menu

Indent line or CTRL+TAB Indent from
selection to next TAB Format menu

Indent according ENTER
to previous line

Move a line or CTRL+SHIFT+TAB Outdent from
selection back one TAB Format menu

Find text F5 Find from
or CTRL+F Edit menu

Find next instance SHIFT+F5 Find Next from
of specified text or CTRL+N Edit menu

Search for and CTRL+F5 Replace from
replace specified text Edit menu

Print current CTRL+P Print from
script File menu

Cut selection CTRL+X Cut from
to Clipboard Edit menu

Copy selection CTRL+C Copy from
to Clipboard Edit menu

Script editor commands for working with text, continued

Key Toolbar Menu
To command button command

Paste from CTRL+V Paste from
Clipboard Edit menu

Import a text file F7 Import File from
into the script File menu

Export selected CTRL+F7 Export Selection
text to a text file from File menu

Write entire script SHIFT+F7 Export Script
to a text file from File menu

Make current line CTRL+HYPHEN Comment from
or selection into Format menu
comments

Uncomment current CTRL+= Uncomment from
line or selection Format menu

Invoke CTRL+D Debug from
Debugger File menu

Use the following commands to save changes to a script.

Script editor commands for saving scripts

Key Toolbar File menu
To command button command

Compile, update, CTRL+S Update Script
and leave the & Exit
current script

Compile and save CTRL+U Update Script
without exiting

Compile, update, CTRL+B Update & Save
and save book Book

Exit without ALT+F4 Exit
saving changes

Check the current CTRL+Y Check Syntax
script’s syntax

Use the following commands to perform additional functions.

Commands for performing additional functions

To Menu command

Change display font Font from View menu

Hide or show toolbar Toolbar from View menu

Hide or show status bar Status Bar from View menu

Checking the syntax of a script

The syntax of a script must be correct before it will run. ToolBook
checks the syntax of a script automatically when you choose Update
Script or Debug from the script editor File menu. You can also check
the syntax of a script as you write it.

To check the syntax of a script:To check the syntax of a script:To check the syntax of a script:To check the syntax of a script:To check the syntax of a script:

1 From the script editor File menu, choose Check Syntax. Or, click
the Check Syntax button on the toolbar.

If ToolBook finds a syntax error, it displays an error message and
highlights the error. Otherwise, it displays the Syntax OK mes-
sage. Assuming a syntax error is found:

2 Click OK to cancel the error message.

ToolBook highlights the first syntax error it finds in the script.

3 Correct the error, and then choose Check Syntax again.

4 Continue checking and correcting the syntax of the script until
ToolBook displays the Syntax OK message.

For details about syntax errors, see Chapter 14, “Debugging
OpenScript programs.”

Working in the Command window
You can use the Command window to execute OpenScript state-
ments without working in a script. You can also use shortcut keys
to move commands or statements from scripts to the Command
window. ToolBook runs statements in the Command window
as soon as you press ENTER. The Command window is available at
both Reader level (if specified in the script of an object) and Author
level, and from the ToolBook Debugger. For details about using the
Command window from the ToolBook Debugger, see Chapter 14,
“Debugging OpenScript programs.”

Use the Command window to test an OpenScript statement or
handler, or to quickly get or set the value of a property, send event
messages to objects, or send messages up the object hierarchy.
Command window messages enter the hierarchy at the page level.

Tip Because you can immediately see what happens when state-
ments are executed, typing commands in the Command window
is a good way to become familiar with OpenScript and test parts
of scripts.

 Figure 2  The Command window

You can use any of the following methods to display the Command
window:

◆ Click the Command Window button on the toolbar.

◆ Choose Command Window from the View menu.

◆ Press SHIFT+F3.

◆ Use the show commandWindow statement in an OpenScript
program.

◆ Use the put command without a target container in an OpenScript
program.

You close the Command window as you close any other window.
You can also move, resize, and maximize the Command window.

Running Command window scripts

ToolBook executes script in the Command window within the
context of the page displayed in the last active viewer (ignoring
“always reader” viewers). This means that:

◆ viewer is the target window. For example, this page refers to the
page displayed in the viewer.

◆ script executes as if it is in the page’s script. For example, self
refers to the page.

The name of the current viewer is displayed in the Command
window title bar.

Using variables in the Command window

By default, variables used in the Command window are untyped
local variables that do not persist beyond the execution of the script
in the Command window. You can, however, declare a variable as
a system variable in the Command window. For the values that
ToolBook assigns to the special variables such as It, my, and self,
see the following table.

Values of special OpenScript containers in the Command
window

Variable Value

It Value set by the last OpenScript command
executed in the Command window

focus ID of object that had the focus at the time the
Command window was displayed

my, self, target uniqueName of currentPage of current
execution context

this page, uniqueName of page, background, or book
background, book displayed in current execution context

CommandWindow *Text entered in the Command window

Tip You can use the Command window to write script for an object
without displaying the script editor. For example, create a button,
select it, and then type the following in the Command window:

script of selection = "to handle buttonClick" & CRLF & \
"go next page" & CRLF & "end"

When you display the script for the button, the script text within the
quotation marks appears.

Using the Command window to run and evaluate
statements

Use the Command window to test commands before making them
part of your scripts. Type an OpenScript statement in the Command
window and then press ENTER. ToolBook executes the statement.

To create multiline statements in the Command window, press
CTRL+ENTER at the end of each line or use a semicolon (;) to sepa-
rate statements.

After you are satisfied with the statements you have entered in the
Command window, you can paste the contents of the Command
window into the script of an object.

Figure 3  Multiline scripts in the Command window



58

Command window history

The Command window keeps a history of the last 20 commands
you entered in the current session. ToolBook displays the most
recent command in the history area above the area where you enter
commands. To resize the history area and show more commands,
drag down the split bar located just below the Command window
title bar. Browse the commands in the history area using either the
mouse or the PAGE UP and PAGE DOWN keys. Clicking a command
in the history area inserts it in the Command window for you to
edit or execute. Double-click a command in the history area to ex-
ecute the command.

Figure 4  Command history area is above the split bar

Debugging scripts in the Command window

To debug a Command window script, place it in the script of an
object, and then debug that script. You cannot use the ToolBook
Debugger to debug scripts in the Command window because
Command window statements are read by ToolBook as one
line of script, which the debugger does not break down into
intermediate lines.

For details about using the ToolBook Debugger, see Chapter 14,
“Debugging OpenScript programs.”

Preventing common scripting errors
You can prevent certain common scripting errors and make the
debugging process easier by following these suggestions:

◆ Use the Command window to execute statements so you can
observe the results and determine which statements work prop-
erly.

◆ Ensure that ToolBook displays the Execution Suspended message
when an execution error occurs by setting the sysSuspend property
to true (the default) while you are debugging a script. For details
about debugging, see Chapter 14, “Debugging OpenScript pro-
grams.”

◆ Name objects consistently and descriptively. For example, if you
have a page that contains a map of the United States, you might
name the page USMap. If another page contains a map of Europe,
you might name it EuropeMap.

◆ Make certain you have named an object before referring to it
by name in a script. Also, enclose names in quotation marks
to speed execution and avoid any confusion with OpenScript key-
words.

◆ Verify that the values for a function referred to in a script, whether
literal or variable, are within the valid range. For example, if you
use the square root function, you cannot specify sqrt(–1) because
negative numbers are outside the valid range for this function.

◆ Do not use OpenScript keywords as variable names.

◆ Make certain that you declare and initialize variables before
referring to them in a script.

◆ Use variable names that describe the variable’s purpose. For
example, if a variable holds the totals from a row in a spreadsheet,
you might name the variable rowTotal. If another variable contains
sales data from Japan, you might name this variable japanSales.

◆ Make sure that variable names are spelled correctly and
onsistently everywhere they occur in scripts. If you misspell the
name of a variable, ToolBook creates a new local variable with
that name.

◆ Include parentheses in or around expressions to force evaluation
in a particular order.

◆ Add comments to your scripts to explain the use of handlers,
statements, and variables so that others can understand how
your script works.

◆ Indent control structures when you write handlers to make the
script easier to read.

◆ When you finish building an application, get rid of any unused
code, old hidden objects, and obsolete user properties.



Chapter 5

Handling messages, events, objects, and properties
This chapter describes the fundamentals of OpenScript programming—
handling messages, events, objects, and properties. Here you will
learn to work with built-in messages, create your own messages, work
with system objects and system properties, create user properties, and
create self-contained objects that are automatically notified of events.

About messages in OpenScript . . . . . . . . . . . . . . . . . . . . . . 62
Working with system objects and properties . . . . . . . . . . . 75
Defining user properties . . . . . . . . . . . . . . . . . . . . . . . . . . . 77
Creating properties using scripts . . . . . . . . . . . . . . . . . . . . . 79
Creating self-contained objects using notify handlers . . . . . 84

About messages in OpenScript
ToolBook has two kinds of messages: built-in and user-defined.
Built-in messages are those ToolBook sends automatically in
response to an event, such as buttonClick, enterField, and newPage.
User-defined messages are those you create and send.

Sending built-in messages

ToolBook sends a built-in message when you perform the action
corresponding to that message. For example, ToolBook sends a
buttonClick message when you click the left mouse button, and
sends a menu-event message when you choose a menu item.
ToolBook sends different types of built-in messages, depending on
what actions you performed, as explained in the following section.

You can also achieve the effect of a built-in message by sending
the message explicitly with the send command. For example, you
can cause a button to respond as if it has been clicked by running
a command such as this one from the Command window:

send buttonClick to button "Next"

Types of built-in messages

You can write handlers for these types of built-in messages:

◆ Menu-event message Sent to the current page when a menu
item is selected from a menu at Reader or Author level. For ex-
ample, if you select an object at Author level and choose Group
Properties from the Object menu, the properties message is sent
to the current page.

◆ Mouse-event message Sent at Reader level when the mouse
button is clicked or the cursor moves over an object. For example,
if a reader clicks a hotword in a field, ToolBook sends
buttonDown, buttonUp, and buttonClick messages to the hotword.

◆ Keyboard-event message Sent at Reader level when a key-
board key is pressed. For example, if a user presses ENTER while
typing in a field, ToolBook sends keyDown, keyChar, and keyUp
messages to that field.

◆ Enter/leave-event message Sent at Reader or Author level
to the viewer, page, background, book, or system when a user
opens or closes an application, book, or viewer; navigates to an-
other page; or clicks a menu. These event messages are also sent
at Reader level to a button, combo box, or field when the
object receives or loses the focus. For example, if a user presses
TAB to move the focus to a field, the enterField message is sent to
the field. If the user clicks another field, ToolBook sends a
leaveField message to the previous field and an enterField message
to the new field. ToolBook also sends a mouseEnter message to
the new field.

◆ Notification message Sent to notify the system of various
events, such as when an object is created, destroyed, or moved.
For example, if you paste a field into a book, ToolBook sends the
make message to the new field. If nothing is happening at Reader
level, the idle message is sent repeatedly to the current page.

◆ Drag-and-drop message Sent before, during, and after a drag-
and-drop operation. ToolBook sends the beginDrag message
to signal the start of the drag operation, and an endDrag message
to indicate that the user has released the mouse button. During
the drag operation, ToolBook continually sends enterDrop,
stillOverDrop, and leaveDrop messages to objects as the cursor
passes over them.

◆ Query message Sent to an object to determine the object’s state
or the value of a particular property. For example, during a drag-
and-drop operation ToolBook sends an allowDrop query message
to each object under the cursor in order to determine whether the
object will accept a drop, which in turn determines whether
ToolBook displays the drag image or the no-drop image.

◆ DDE-event message Sent at Reader or Author level when
information is exchanged between applications through dynamic
data exchange (DDE). For example, if an instance of Microsoft Ex-
cel requests data from ToolBook, the remoteGet message is sent
to the current page.

◆ Standard message Sent at Reader level only to the ToolBook
system (not up the object hierarchy). For example, editScript.

? For a complete list of messages, refer to the OpenScript reference in
online Help.

Sending user-defined messages

User-defined messages allow you to create scripts for specific tasks
that are not already part of ToolBook. Sending a user-defined
message is like calling one of your own procedures or functions in
another programming language.

You can send a user-defined message using the send command.
You can also create menu items that send a user-defined message.
In addition to creating the user-defined message, you must write
a handler that responds to it.

For example, you might create an index page in your application,
and then add a menu item named Index to the Page menu. ToolBook
automatically sends the message index to the page when a user

chooses this menu item, or when you execute the command send
index in a script. To define the response to the index message,
write a handler like the following:

--Place this handler in the page script, the background script,
--or the book script
to handle index

go to page "Index" of book "overview.tbk"
end

If you don’t write a handler for a user-defined message, ToolBook
displays an error message:

◆ For menu-event messages related to a user-defined menu item,
ToolBook displays a message that there is no handler for the
user-defined message.

◆ For other user-defined messages, ToolBook displays the Execution
Suspended message.

When you create a user-defined message, make sure its name
includes only letters and the underscore character, and that it is
not an OpenScript keyword.

? For details on keywords, refer to the OpenScript reference in online
Help.

User-defined messages are particularly effective for creating
modular applications because you can create individual handlers
that are easier to code and maintain. For example, to streamline an
enterBook handler, you can create separate handlers for different
tasks, such as initializing menus, sizing windows, and establishing
default values. Then, you can send user-defined messages from the
enterBook handler to run these handlers, rather than including all of
the statements required to do these tasks. The individual handlers

will run automatically when the book is opened, or you can call
them as needed with a send command.

to handle enterBook
forward
--The following are all user-defined messages
send initMenus
send sizeAppWindow
send setSystemDefaults

end

Placing handlers for built-in messages

You can place a handler for a built-in message in the target object’s
script or anywhere higher in the object hierarchy. Use these tips to
help you decide where in the object hierarchy to place the handler
for a specific message:

◆ Place handlers for built-in messages in the script of the target
object or higher. For example, because leavePage is automatically
sent to the page, don’t put a leavePage handler in the script of a
button.

◆ To program general behavior for several objects, place a message
handler high in the object hierarchy. For example, if the handler
defines behavior for several objects on a page, put the handler in
the script for that page, or group the objects and place the handler
in the script of the group. This strategy allows you to change, add,
or remove the component items of the page or group without
changing the scripts of the individual objects.

For example:

--Handler that asks for confirmation each time the user
--navigates to a new page; place this handler in a book script
to handle next

request "Are you sure you want to proceed?" with "Yes" or "No"
if It is "Yes" then

forward
end if

end next

◆ To program specific behavior for one object, or to cause various
objects to respond differently to the same message, put the
handler in the script of the target object. Or, put the handler in
the page or book script, and then use the target property to
distinguish which object received the message using a control
structure.

For example:

--Handles navigation for buttons named "Next" and "Previous"
--and for hotwords whose name matches that of a page
to handle buttonClick

conditions
when object of target is "button"

send (name of target)
when object of target is "hotword"

go to page (name of target)
else

forward --User clicked some other object
end conditions

end buttonClick

◆ To define both specific and general behavior for an object, put a
handler in the target object’s script to define specific behavior and
forward the message. Then, put a handler for the same message
higher in the message hierarchy to define general behavior. To
allow a message to get to the ToolBook system while bypassing
all scripts higher in the message hierarchy, use forward to system.

Note You can also create self-contained objects that are notified
automatically of messages that are forwarded to the page they are
on. For details, see “Creating self-contained objects using notify
handlers” later in this chapter.

Sending messages from scripts

Use the send or forward command to send a message regardless
of whether the event that usually triggers the message occurred.
By sending messages from scripts, you can:

◆ run a handler for a user-defined message.

◆ execute scripts of hidden objects.

◆ handle a message more than once.

◆ send menu-event messages without choosing the menu item.

◆ send mouse or keyboard messages at Author level.

◆ send enter- and leave-event messages (such as enterBook or
enterSystem) without actually navigating.

◆ streamline handlers so they are easier to maintain.

Use the send command to run a handler in the current object’s script
or send a message to any other object, including objects lower in the
hierarchy or in another book. For example, send buttonDoubleClick to

selection runs the corresponding handler in the script of the selected
object, and send buttonClick to this page runs the corresponding
handler in the page script. When you send a message with the send
command, ToolBook sets the target property to the uniqueName of
the object you sent the message to.

Use the forward command to pass the current message to the next
highest object in the hierarchy. For example, the forward command
in an enterBook handler in a page script sends the enterBook mes-
sage to the background. Use forward to system to forward a mes-
sage directly to the ToolBook system, skipping any other
objects in the object hierarchy. Forwarding a message does not
change the target property.

A handler with a send or forward statement that runs another handler
is referred to as a calling handler. The handler that is invoked
is the called handler. When ToolBook encounters a send or
forward statement in a script, it suspends the calling handler until
the called handler is finished, and any additional handlers that the
called handler in turn calls. Eventually, control returns to the calling
handler.

Tip Sending a message to a specific object is faster than allowing
the message to travel up the hierarchy.

Sending message names as variables

You can pass the name of a message as a variable. To force
ToolBook to recognize that you want the contents of the variable
sent, not the variable name, put parentheses around the name of
the variable that contains the message to be sent. For example:

ask "What message do you want to send?"
send (It)

Changing or disabling built-in messages

Because many messages move up the object hierarchy before they
reach the ToolBook system, you can change a message or disable it
altogether. For example, you can trap a keyboard-event message by
writing a handler for it, and if you do not forward the message,
ToolBook doesn’t ever receive and react to the message. For
example:

◆ You can trap the author message and display a message instead
of actually switching to Author level:

--Traps the author message and asks for a password before
--changing to Author level
to  handle author

ask password "Enter password for Author level:"
if It is "uAKmnKpw" then --The encrypted form of "ToolBook"

forward
end if

end  author

◆ You can trap the right mouse click and disable that event:

--Traps and disables the rightButtonDoubleClick message
--(no forward here, so the handler stops the message)
to handle rightButtonDoubleClick
end

◆ You can trap keyboard events and substitute a character
(for example, an uppercase letter for a lowercase one) before
forwarding the message:

to handle keyChar key, isShift, isCtrl
--The  uppercase( ) function requires a letter, but key values
--are passed as numbers, so they must be converted
uppercaseKey = uppercase(ansiToChar( key ))

key = charToAnsi(uppercaseKey)
forward --To display the character

end

Using parameters with messages

ToolBook sends some messages with parameters that contain
additional information about the message. For example, ToolBook
includes a parameter for the mouse position with the built-in message
buttonClick. You can add parameter variables to handlers to capture
this additional information. For example:

-- "Position" is a parameter variable containing the location
--  of the cursor when the button was clicked
to handle buttonClick pos

request "You clicked at " && pos
end

When you send a built-in or user-defined message with the send
command, you can send values or variables as parameters for the
message. Parameters are always placed after the message name
and multiple parameters are separated with commas:

--Sends a user-defined message with 3 parameters
send specialEffects location, color, grayTone

In the handler for the message, put the parameter variables in
the same order, so you have access to the values that were passed:

--Handler for a user-defined message with three parameters
to handle specialEffects location, color, grayTone

Naming variables for parameters

Parameters are placeholders, so only their position in the to handle
statement is important. When you create a parameter variable, you
can give it any valid variable name—it need not have the same name
as the parameter in the calling routine, only the same position. For
example, the first statement works as well as the second:

to handle specialEffects location, color, grayTone

to handle specialEffects place, colorCode, gradation

Scope of parameters

A parameter is passed by value—that is, it is a copy of the value in
the calling handler. Therefore, parameters become local variables
in the handler to which they are sent. If you change the value of a
parameter in the called handler, it does not affect the value of the
corresponding parameter in the calling handler.

Declaring data types for parameters

You can add data-type declarations to parameters in a to handle
statement. As with any variable for which you declare a data type,
this can make the parameter more efficient because it requires less
memory and ToolBook can access the data faster. When you declare
a data type for a variable, ToolBook converts the parameter
variables to the appropriate type, and enforces that data type for the
rest of the handler. For example:

to handle buttonClick point loc, logical isShift, logical isCtrl
loc = mousePosition of this window
loc = "abc" --Error (loc has been typed as a point, not string)
--Further statements here

end



73

For details about data types, see Chapter 7, “Using variables, values,
and text.”

Counting and listing parameters

Besides referring to a parameter by its name, you can refer to a
parameter in several other ways:

◆ Use the special variable argList to get a list of the parameters
passed, and then refer to a parameter as an item in argList:

put item 3 of argList into field "MyList" of this page

◆ Use the special variable argCount to get the number of parameters
passed to determine (for example) that all of the parameters you
intended to pass were passed.

◆ Use itemCount(argList) to get the number of items passed with
a message, which is different from using argCount. For example,
if a message has a location parameter, the parameter will contain
a list of two numbers. In this case, itemCount(argList) returns 2,
but the value of argCount is 1.

Working with system objects and properties
In addition to the objects that you create, such as buttons, fields,
pages, and backgrounds, ToolBook contains a number of system ob-
jects that belong to the ToolBook system, but which you can ma-
nipulate to change the way ToolBook looks and behaves.
Examples of system objects are the Command window, tool palette,
and polygon palette.

ToolBook system objects differ from objects such as buttons and
pages because you cannot create or destroy system objects. However,
they have properties you can get or set. For example, you can hide
and show these objects, set the caption of the status bar, and so
forth. The object type names for the ToolBook system objects are:

colorTray patternPalette statusControls
commandWindow polygonPalette statusIndicators
lineEndsPalette statusBar toolBar
linePalette statusBox toolPalette

Note Only the statusBar system property is available in the run-time
version of ToolBook.

System properties define default behavior for the entire ToolBook
system or identify objects that are currently receiving messages. By
changing the values of certain system properties you can set default
properties for new objects, set margins and other printer properties,
or format system-wide values such as the current date and time.

For example, if you set the system property sysFontStyle to bold,
the text in new fields and buttons is automatically bold. By getting
the value of other system properties such as target, focus, and selection,
you can determine which object is the target for user events.

There are five types of system properties:

◆ General system properties These properties affect every book
opened during the current ToolBook instance. For example, if you
set the sysTimeFormat property in the enterBook handler for the
first book opened in a ToolBook instance, that format will be used
in all other books opened in that instance until another statement
resets the format.

◆ Event-focus properties The target property indicates the object
that receives the current message. The targetWindow property
indicates the viewer that displays that object. The focusWindow
property indicates the viewer that is currently active (which may or
may not be the same as targetWindow).

◆ Printer system properties These properties affect various
aspects of how a book prints. For example, if you set the
printerBorders property to true, all pages and reports printed from
that book will have a border around them.

◆ International system properties These properties can be used
to automatically adapt books to the local conventions of
specific countries. For example, the value of sysDecimal specifies
which character separates the fractional part of formatted
numbers.

◆ Startup system properties These properties affect all books
in all instances of ToolBook. When you set a startup system prop-
erty, ToolBook changes a value in the Tb90.ini file.

? For details about the properties you can get and set for a specific
system object, and for a list of all system properties, refer to the
OpenScript reference in online Help.

Defining user properties
You can assign a user property to an object to extend the kind of
data the object can contain. For example, you can create a lastScore
property for a book to store the final score in a game application.
You can use user properties as global containers, much like system
variables. However, unlike system variables, user properties are
retained when you close the book and are associated with only one
object.

Tip Data stored in user properties takes up space in a book; there-
fore, you should not include a lot of user properties if disk space is
a consideration in your application.

Creating a user property

You create a user property by using it in an OpenScript statement as
you would any built-in property. If the property you name does not
already exist, ToolBook automatically creates it and sets it to the
value specified:

userScore of button "Play" = 0

Note: If you misspell the name of a built-in property when setting
its value (for example, set fullColor to blue instead of set fillColor to
blue), ToolBook will create a user property instead of reporting
an error.

When you refer to a user property, you must always include a
reference to the object that owns the property. If you omit the object
reference, ToolBook creates a local variable instead of a property.
For example:

userScore of this book = 190 --Defines a user property
userScore = 190 --Defines a local variable

Getting the value of a user property

You get the value of a user property the same way you get the value
of a built-in property. For example, the following statement gets the
value of a userScore property:

get userScore of button "Play" --Puts value into It

Listing user properties

You can use the keyword userProperties to list the properties you
have created for any object. The most recently assigned user property
appears at the front of the list. For example:

--Puts list of user properties into the Command window
put userProperties of button "Play"

Deleting a user property

To delete a user property and remove its name from the list
userProperties, you can set the property value to null or use the clear
command.

For example, to set a user property value to null:

userScore of button “Play” = null

To clear a user property:

clear userScore of button “Play”

Note To recover the space used by a user property that has been
deleted, choose Save As from the File menu or run the OpenScript
save as command and give the book a new name. This reorganizes
the book to its most efficient size.

Creating properties using scripts
You can create properties with values determined by the handlers
you write. The handlers are called when you use the set or get
commands, just as if you were setting or getting a built-in or user
property. However, instead of setting or getting a property value,
ToolBook calls your handler, which can set values, modify objects,
or run any other process that your application requires. Using this
technique, you can create a property that:

◆ filters or converts characters, such as a ucText property that
returns the uppercase version of the text of a field.

◆ checks for specific values when you assign them to an object,
such as a dateText property that sets the text of a field but only
allows valid dates.

◆ sets a group of properties to a predetermined value, such
as a property that resets a number of objects to a single color,
or a menu to a default configuration.

◆ calculates a value using DDE calls to another Windows
application.

◆ adjusts visual elements of the application as its value changes,
such as the bars on a chart.

To create a property using a script, write a to set handler. When
you use a set command to assign a value to a property, ToolBook
determines if the property is built-in. If so, ToolBook assigns the
value to that property. However, if the property is not built-in,
ToolBook sends a message to the target object and then up the ob-
ject hierarchy. ToolBook runs the to set handler that matches
the property name. If there is no corresponding to set handler,
ToolBook creates a user property and assigns a value to it.

Creating a to set handler

The syntax for a to set handler is:

to set <propertyName> [parameters] to <value>
--Further statements here

end

When you call the handler with a set command, the value that
you assign to the property is passed to the to set handler as the
<value> parameter. For example, the following to set handler defines
a property called texture for a filled object (such as an ellipse) or for
the background. Possible values for the texture property are bricks,
waves, hearts, or none. Based on the value you assign to the texture
propery, the to set handler changes the pattern and fill color of the
target object to a new value.

to set texture to vTextureName
validObjects = "ellipse,rectangle,polygon,pie," & \

"background,irregularPolygon,roundedRectangle"
--Check that this is a valid object type
if object of target is not in validObjects

break
end if
conditions

when vTextureName is "bricks"
pattern of target = 105
fillColor of target = red

when vTextureName is "waves"
pattern of target = 94
fillColor of target = blue

when vTextureName is "hearts"
pattern of target = 114

fillColor of target = 300,50,100 --Pink
when vTextureName is "none"

pattern of target = 254 --None
fillColor of target = white

end conditions
end texture

To call this handler, use a set command such as:

texture of ellipse “circle” = “bricks”

Placing to set handlers in the object hierarchy

You can place to set handlers in the script of a target object or
anywhere higher in the object hierarchy. Use the same guidelines
for placing the to set handler as you use for other handlers—for
example, if you are creating a general-purpose handler, place it in
the script of a book or system book. You can identify the object for
which you are assigning the property using the target property.

Note When you write a to set or to get handler for a property, do
not refer to that property in the body of the handler. If a statement
in the handler tries to set or get the same property,
ToolBook calls the handler again, creating an infinite loop.

Getting a property’s value with a to get handler
To define the value of a property created with a to set handler,
create a corresponding to get handler. The syntax for a to get
handler is:

to get <propertyName> [parameters]
--(calculate value here)
return <value>

end

The following example shows the to get handler that corresponds
to the to set texture handler shown earlier:

to get texture
validObjects = “ellipse,rectangle,polygon,pie,” & \

“background,irregularPolygon,roundedRectangle”
--Check that this is a valid object type
if object of target is not in validObjects

break
end if
conditions

when pattern of target = 105 and fillColor of target = red
retVal = “bricks”

when pattern of target = 94 and fillColor of target = blue
retVal = “waves”

when pattern of target = 144 and fillColor of target = 300,50,100 --
Pink

retVal = “hearts”
when pattern of target = 254 and fillColor of target = white

retVal = “none”
else

retVal = pattern of target & “,” & fillColor of target
end conditions
return retVal

end texture

As with a built-in property, ToolBook will invoke the to get handler
whenever you get the value of the property. For example:

request texture of ellipse "circle"

or

text of field “display” = texture of ellipse “circle”

If there is no to get handler for a property created with a to set
handler, ToolBook returns null. ToolBook searches the hierarchy for a to
get handler the same way it searches for a to set handler, so it’s a good
idea to keep these together in the same script.

Tip Remember that if you provide to get and to set handlers, ToolBook
will call the to get handler whenever it needs to get the property, and the
to set handler whenever it needs to set the property. This happens in situ-
ations that are not immediately obvious. For example, consider a bright-
ness property that changes the brightness of an object:

to get brightness
return item 2 of fillColor of target

end
to set brightness to value

--maximum brightness is 100, minimum is 0
if value > 100

value = 100
else

if value < 0
value = 0

end
end
item 2 of fillColor of target = value

end

If you execute a command such as:

increment brightness of ellipse “light”

ToolBook will call the to get handler to get the value of brightness, incre-
ment it by one, and then call the to set handler to set it again.

Note You can also use to get handlers to create user-defined
functions. For details, see Chapter 6, “Creating an OpenScript statement.”

Creating self-contained objects using notify handlers
A notify handler is one that is automatically notified when the
message it handles reaches the page. Usually, objects receive
messages only if they are the targets of a user action or of the send
command, or if another handler forwards the message to them.
However, if an object contains a notify handler, the page sends
a copy of the message to that object. For example, you can create
a “Next” button that is automatically notified of any enterPage event.
If the user navigates to the last page of the book, a notify handler
in the button can disable the button.

The advantage of using a notify handler is that rather than simply
responding to a message, the object can be completely self-con-
tained—you do not have to add a handler elsewhere to send
a message to the object. You can copy the object to another book
with no other changes. In the example of the “Next” button, without
notification you would have to change the book script to send an
enterPage message to the button in addition to simply copying the
button to the background.

Types of notify handlers

You can use two types of notify handlers: notifyBefore handlers and
notifyAfter handlers. NotifyBefore handlers execute just before the
message for which they are being notified is sent, and notifyAfter
handlers execute just after the message is sent. You write a notify
handler just as you write a normal handler, substituting the words
notifyBefore or notifyAfter for the words to handle in the first line
of the handler:

notifyBefore enterPage
--Further statements here

end

As an example, you can create an on-screen clock by displaying the
time of day in a field. Create a notifyBefore handler in the field that
responds to the idle event by setting the field’s text to the current
time:

notifyBefore idle
system oldTime
if oldTime is not sysTime then --Update if time has changed

my text = sysTime
oldTime = sysTime

end if
end idle

Another use for a notify handler is to create an object that is visible
only at Author level. The following script illustrates how you can use
notify handlers to ensure that the object is always hidden when the
user switches to Reader level:

notifyBefore reader
hide self

end
notifyAfter author

show self
end
notifyBefore enterPage

if sysLevel = "author"
show self

else
hide self

end if
end enterPage

Using notify handlers

When you use notify handlers, keep these points in mind:

◆ Multiple objects can be notified of the same message if each
object contains a notify handler. However, you cannot specify the
order in which the objects are notified of the message.

◆ Do not use the forward command in notify handlers. If you do,
ToolBook generates an error when you attempt to save the script.

◆ To stop a notify handler from processing a message in its script,
use break.

◆ You can put notify handlers into the script of any object on a page
or background. However, you cannot put notify handlers into the
scripts of pages, backgrounds, viewers, or books.

◆ Notify handlers respond only to messages that reach a page. If
an object traps a message and does not forward it to the page,
ToolBook will not call the notify handler. Similarly, if the
message goes first to an object on the background and the script
in that object does not explicitly send it to the page, ToolBook will
not call the notify handler.

The table on the next page shows the properties and OpenScript
commands that you can use with notify handlers.

Properties and commands for notify handlers

Keyword Description

notifyObjects Page or background property that specifies
a list of all objects that contain notify handlers

notifyAfterMessages, Properties of an object that list the notify
notifyBeforeMessages handlers written for that object

sendNotifyAfter Command that activates a specific notifyAfter
handler within a script

sendNotifyBefore Command that activates a specific notifyBefore
handler within a script

Chapter 6

Creating an OpenScript statement
This chapter describes how to create a statement in OpenScript and
includes detailed information about the components that make up
a statement: commands, functions, expressions, and operators. You
will also learn about control structures, such as if/then/else and do/
until, which you use to control blocks of statements in a script and
expressions.

About OpenScript commands . . . . . . . . . . . . . . . . . . . . . . . 90
Using OpenScript control structures . . . . . . . . . . . . . . . . . . 92
Using functions . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 100
Creating user-defined functions . . . . . . . . . . . . . . . . . . . . 102
Creating expressions with operators . . . . . . . . . . . . . . . . . 104

About OpenScript commands
To create an OpenScript statement, use a command and the
parameters the command requires. OpenScript includes commands
to:

◆ declare variables and get and set the values of variables or
properties (get, increment, local, pop, push, put, set, and system).

◆ manipulate objects (align, clear, draw, and move).

◆ interact with the user and navigate between pages (ask, go,
request, search, sort, and transition).

◆ print, save, read, or write data in ToolBook files (export, openFile,
print, readFile, and writeFile).

◆ add and remove menu items (addMenu and addMenuItem).

◆ exchange data with other Windows applications (executeRemote).

◆ extend programming code to include routines and data within
Windows DLLs.

? For a complete list of commands and information on them, refer
to the OpenScript reference in online Help.

Using commands versus sending messages

In some instances, you can either use a command or send a message
to accomplish the same task. For example, commands such as align,
clear, export, import, run, save as, search, and sort are also messages.
Use these guidelines to determine when to use a command or
message:

◆ Use commands such as export or search to execute the action
without displaying the related dialog box. Use commands with
keywords such as author or reader if you want to avoid running
related message handlers. For example, the statement set sysLevel
to reader does not execute actions in a to handle reader handler,
but the working level is changed to Reader.

◆ Send messages such as import or saveAs if you want the user to
perform the action by filling in a dialog box. Send a message such
as reader or clear when you want any handlers for the message
to run in addition to accomplishing the primary action. Sending
a message results in the same behavior as choosing the related
menu item.

Using OpenScript control structures
A control structure is a group of statements that runs under particular
circumstances, such as when handling a message, looping (repeating)
a block of statements, or testing a condition. The OpenScript control
structures are shown below.

OpenScript control structures

Type Commands Description

Looping step Repeatsa fixed number
of times

do/until Repeats until a condition
is met; always executed at
least once

while Repeats while a condition
is true

Establishing if/then/else Executes statements if a
conditions condition is true; else branch

optional

conditions Executes one from a set of
mutually exclusive choices

Handling to handle Reacts to a system or user
messages message

Creating to set Defines a script to create
properties and or calculate a value for a
user-defined property or user-defined
functions function

to get Defines a script to return
the value of a user-defined
function or calculated user
property

(continued)

OpenScript control structures, continued

Type Commands Description

Setting target in <viewer> Temporarily makes a viewer
window the target window for

running scripts

Printing start spooler Routes subsequent print
(spooling) commandsto the current

print device

Linking DLLs linkDLL Declares functions from
a DLL to be called

Windows translateWindowMessage32 Establishes handling of
message Windows-to-OpenScript

message translation

You always begin a control structure with one of the control
structures from the table, and close with the word end. The only
exception is do, which ends with until <expression>. In between you
can put any number of OpenScript statements. You can repeat the
name of the control structure after end to make it easier to see
where a specific control structure ends, as in this example:

step ctr from 1 to pageCount of this book
flip

end step

Examples of control structures

The most common control structures, such as to handle and
if/then/else, are extensively illustrated throughout the documenta-
tion. This section shows you how to use some of the less common

control structures, such as conditions, do/until, in <viewer>, linkDLL,
and while.

For details about to set and to get handlers, see Chapter 5, “Handling
messages, events, objects, and properties.” For details about user-
defined functions, see “Creating user-defined functions” later in
this chapter. For details about translating Windows messages, see
Chapter 13, “Using dynamic-link libraries.”

Looping and testing conditions

Use a looping control structure to repeat a series of statements.
The while control structure tests at the beginning of the loop. By
contrast, the do control structure tests at the end, so the loop
always executes at least one time. Use the step control structure
when you know in advance how many times you want to repeat
the statements.

--Uses the while control structure to get just the file name
--from a string that contains a path plus the file name
to get nameOnly fFile

vOffset = offset("\",fFile)
while vOffset > 0

clear chars 1 to vOffset of fFile
vOffset = offset("\",fFile)

end while --Repeats control structure name for clarity
return fFile

end nameOnly

--Uses the do control structure to display the cursor
--position in the status bar at Reader level
to handle buttonClick

if visible of statusBar is false then
show statusBar

(continued)

end if
do

caption of statusBar = mousePosition of this window
until keyState(keyLeftButton) is up

end buttonClick

There are two ways to test conditions in OpenScript. For a simple
test, use if/then, adding an else branch if you want to explictly define
the statements that should execute otherwise. To test for one of a
series of mutually exclusive possibilities, use the conditions/when/
else control structure. ToolBook tests each condition in turn,
executing the statements that follow the first condition that is true.
As with an if/then control structure, you can add an else branch to the
conditions to indicate the statements that should execute otherwise.

--Uses the conditions control structure to test for several
--possible user responses
to handle buttonClick

request "Do you want to proceed?" with "Continue" or "Quit" \
or "Restart"

conditions
when It is "Continue"

request "You chose Continue"
when It is "Quit"

request "You chose Quit"
when It is "Restart"

request "You chose Restart"
else

request "You closed without choosing anything"
end conditions

end buttonClick

Setting the target window

If you are running a script in one viewer but want to affect objects
in another, you can use the in control structure to temporarily make
the second viewer the target window. The statements inside the
control structure then run as if they were running in the second
viewer, and implicit object references and the generic references this
and my refer to the second viewer. For example, you can use in if
you want a user to click a button in the current viewer to navigate
to a particular page in another viewer.

to handle buttonClick
show viewer "Topics"
in viewer "Topics"

currentPage = "Contents"
end in

end buttonClick

Spooling

Use the start spooler control structure to initialize the print spooler.
The print and print eject commands work within this control structure
only, and so you must use it to print from a script. For example:

to handle buttonClick
start spooler --Encloses print statements

print all pages
end spooler

end buttonClick

Linking to DLLs

You use the linkDLL control structure to initialize and declare a link
to a DLL.

-- Link a DLL to call a function that returns a file size
to handle buttonClick

linkDLL "draw.dll"
LONG getPenSize(STRING)

end linkDLL
  request getPenSize("blue")
end buttonClick

To determine if a DLL is already linked, you can check the value
of the system property sysLinkedDLLs, which contains a list of
the DLLs currently linked. For details about linking DLLs, see Chap-
ter 13, “Using dynamic-link libraries.”

Nesting control structures

You can nest all control structures within other control structures;
however, you cannot put a start spooler control structure within
another start spooler structure. You can nest control structures any
number of levels deep.

Interrupting control structures

To interrupt or modify the flow of a control structure, use break
and continue inside the structure. Use the break statement to quit
a handler without finishing all of the statements in it, or in loops
to quit the loop early, such as when an error is detected.

If you use break to quit the handler, control returns to whatever pro-

cess called the current handler. The break to system statement stops
execution of all handlers currently running, including any that called
the current handler. The following handler uses the break statement
to prevent the handler from completing and exiting ToolBook:

to handle buttonClick
request "Do you want to quit?" with "Continue" or "Quit"
if It is "Continue" then

request "You have elected to continue."
break --Quits handler

end if
save changes to this book
send exit --Exits from ToolBook

end buttonClick

To use break to quit a control structure within a handler (such
as step or do), include the name of the control structure after
the word break. For example:

step i from 1 to 1000
--Checks room available on page
if percentFreeSpace of this page < 10 then

break step
else
… --Further statements here
end  if --Otherwise continue processing

end step --Handler resumes here after break statement
--or if step has finished successfully

The continue statement is used to skip statements in a control
structure (most often step or do/until) and continue directly to the
next iteration. For example:

--This handler counts the objects on a page that are not buttons
to handle buttonClick

vTotalItems = 0
--Gets a list of all the objects on the page
itemList = objects of this page
step i from 1 to itemCount(itemList)

--Gets the next object from the list of objects
pop itemList
--Tests to see if the current item is a button
if object of It is "button" then

--Skips to the "end step" statement
continue

end if
--Otherwise increments the total object count
increment vTotalItems

end step
request "total items = " & vTotalItems

end buttonClick

Using functions
Functions are procedures that you use in your script to calculate
a value or perform an operation. ToolBook has over 200 built-in
functions for arithmetic, statistics, string handling, financial calcu-
lations, and other purposes. For example:

--The function "textlineCount" returns the number of lines in a field
headCount = textlineCount(text of field "Attendees")

--The function "keyState" returns "up" or "down" for the
--key value that you request
if keyState(keyShift) is down then

… --Further statements here
end

? For more information on functions, refer to the OpenScript reference

in online Help.

You can also write your own user-defined functions. For details, see
“Creating user-defined functions” later in this chapter.

Using parameters

When you call a function, you include parameters for the values
used to calculate the return value. For example, the variable vName
is the parameter for the uppercase function in this script fragment:

vName = “Mary Jones”
get uppercase(vName) —Puts MARY JONES into It

Note that the variable vName still contains the mixed-case “Mary
Jones” text.

Except for array parameters that use the by reference special term,
OpenScript passes parameters by value, copying the value when you
pass it in a function. The values of the parameters in the calling
handler do not change.

Passing parameters by value can use up a lot of memory, especially
when the variable is a large array. This is because ToolBook makes
a copy of the variable to use within the function. If you want to pass
variables by reference that are not arrays, you must allocate memory
and pass a pointer to the memory as a parameter to your function.

For details about using DLL functions, see Chapter 13, “Using dy-
namic-link libraries.”

? For information on passing arrays by reference, see the entry for
by reference in the Openscript reference in online Help.

Creating user-defined functions
A user-defined function is a handler you write that calculates a
value or runs a procedure and then returns a result. For example, to
calculate the same value often, such as cost of goods, you can write
the calculation as a user-defined function, and then call the function
as needed. User-defined functions are particularly useful when the
calculation is complex or when you need to perform the same
calculation in different applications.

You can use user-defined functions anywhere that you can use built-
in functions. To create a user-defined function, write a to get han-
dler. The handler returns the calculated value using the return
statement. For example:

--Defines function name and parameters
to get grossMargin Sales, COGS

--Calculates and returns the results to the calling handler
return Sales - COGS

end

You can use several statements of varying kind in a to get handler—
commands, messages sent to run other handlers, or other function
calls. The to get handler must include a return statement that
specifies the value to be returned. The return statement acts as a
break for the function—no statements following a return statement
are executed.

Naming user-defined functions

You can assign a name to a user-defined function that is similar
to (but not exactly like) the name of a built-in function. Names you
assign to a user-defined function:

◆ must begin with a letter, an underscore character (_), or a prefix
symbol, such as the at symbol (@).

◆ cannot contain spaces or any characters other than letters,
numbers, the underscore character, or the prefix symbol.

The following example uses the at symbol (@) to distinguish a user-
defined function from a built-in function.

--Calls the built-in function named average
request average(1,2,3,4,5)

--Calls the user-defined function named average
request @average(1,2,3,4,5)

Placing user-defined functions
in the object hierarchy

You can place a to get handler anywhere in the object hierarchy.
However, it’s good practice to define user-defined functions in
the book script because it makes the scripts easier to maintain. If
you define functions in the book script of a system book, you can
use them with any application and avoid repeating the function
definitions in other books. However, the slight overhead of search-
ing up the hierarchy for the function might affect the performance
of an application.

Calling user-defined functions

After you define a function, you can call it from any handler the same
way you call a built-in function. ToolBook searches the hierarchy for
to get handlers as for any message handlers. The following example
calls the user-defined function grossMargin( ) created earlier:

to handle buttonClick
local vSales, vCOGS
vSales = text of field "Total Sales"
vCOGS = text of field "Unit Cost" + text of field "Overhead"
text of field "Gross Margin" of page 3 = grossMargin(vSales,vCOGS)

end

Creating expressions with operators
An expression is any combination of a formula, calculation, com-
parison, or set of values that results in a single value. You create an
expression by linking its elements with an operator, which is a word
or symbol that indicates the operation to be performed. For example:

--Multiplies two values
subTotal = vItems * vQty

--Joins three strings into one
greeting = "Dear" && title && lastName

--Divides the values of two functions
average = sum(allItems) / itemCount(allItems)

--Compares two values
if text of field "Amount" > vLimit then …

You can compose expressions from any combination of object
references, properties, references to strings or to variables, numbers,

literal values, and other expressions, including:

◆ containers Anything that holds a value, such as variables,
properties, the text of a field, or the Command window.

◆ functions References that call a predefined operation and return
a value. For details, see “Using functions” earlier in this chapter.

◆ literal values Values entered character by character in a
statement. For example:

text of field "Zip" = "Please enter a 9-digit zip code."

◆ constants Specific values referred to by name, such as pi, which
refers to the value 3.141592653589793.

◆ expressions with operators A combination of operators and
operands that results in a value. For example:

--Results in 250 if the current value of svCost is 100
100 + svCost + 50

◆ expressions with strings References to parts of strings. For
example, if the text of field ABC contains primary colors, then the
following statement refers to the value primary:

word 1 of text of field "ABC"

The value returned by an expression can be a number, character
string, or logical value (true or false).

Operators in ToolBook

You can create expressions using four types of operators: arithmetic,
logical, string, or bitwise, as shown in the following tables.

Arithmetic operators

Use arithmetic operators to create mathematical expressions.

Operators for arithmetic expressions

Operator Meaning

+ Addition. For example:
It = 25.3 + vFirst --Puts result into It

– Subtraction. For example:
It = 25 - 4 --Puts 21 into It

* Multiplication. For example:
It = 25 * 4 --Puts 100 into It

/ Division. For example:
It = 25 / 4 --Puts 6.25 into It

div Division (returns only the integer). For example:
It = 25 div 4.5 --Puts 5 into It

– Negation. For example:
It = -4 + 25 --Puts 21 into It

^ Exponentiation. For example:
It = 2 ^ 4 --Puts 16 into It

mod Modulo (returns the remainder). For example:
It = 16 mod 5.0 --Puts 1 into It

Logical operators

Use a logical operator to evaluate two expressions and return true
or false. Expressions containing dates, currency, or other formatted
data can be compared as <type> to force ToolBook to convert the
data in the comparison to the appropriate format first.

Operators for logical comparisons

Operator Meaning

= or is Equal. True if both expressions have same value.
For example:
It = (25=4) --Puts false into It

if color is blue then
color = red

end

< Less than. True if expression on left is less than
expression on right. For example:
if balance < lowLimit then

text of field "odWarning" = "Overdraft"
end

<= Less than or equal to. True if expression on left is less
than or equal to expression on right. For example:
if text of field "Date" <= sysDate as date

request "Too early to process"
end

> Greater than. True if expression on left is greater than
expression on right. For example:
get 25 > 4 --Puts true into It

Operators for logical comparisons, continued

Operator Meaning

>= Greater than or equal to. True if expression on
left is greater than or equal to expression on right.
For example:
if vFirst >= vLast

request "Last item is less than first."
end

<> or is not Not equal to. True if two expressions are different.
For example:
if svText <> text of field "Text"

text of field "Text" = svText
end

and Boolean And. True if both expressions are true.
For example:
if text of field "input" > "01/01/94" as date and \

text of field "refAmount" < 25
request "The expression is true."

end

or Boolean Or. True if either expression is true.
For example:
if interest > 10.5 or interest < 9.5 then

request "Interest rate is out of range."
end

contains String search. True if expression on right is found in
expression on left. For example:
if text of field "emperors" contains "Roman"

go to book "c:\lessons\roman.tbk"
end

(continued)

Operators for logical comparisons, continued

Operator Meaning

is in String search. True if expression on left is found in
expression on right. For example:
if nameOnly(vName) is in text of field \

"chooseIndex"
request "That index is already open."

end

is not in String search. True if expression on left is not found in
expression on right. For example:
if "Nero" is not in vAnswerText

request "Please try again."
end

not Not. True if expression on right is false; false if expression
on right is true. For example:
--File name is in title bar
if not captionShown of mainWindow

request "Do you want to personalize this book?" \
with "Yes" and "No"

if It is "Yes"
send personalize -User-defined message

end if
end if

String comparisons are not case-sensitive, so b is equal to B. If
you want to compare values by case, first use the charToAnsi( )
function to convert the values to their ANSI equivalents. The fol-
lowing is a user-defined function that performs a case-sensitive

(continued)

comparison on two strings:

to get compareCase string1, string2
--If a case-insensitive check fails, the strings are different
if string1 <> string2 then

return false --Return to calling handler
end if
--Compare character-by-character
step ctr from 1 to charCount(string1)

if charToAnsi(char ctr of string1) <> \
charToAnsi(char ctr of string2) then

return false --Return to calling handler
end if

end step
return true

end compareCase

Text values in comparison expressions are converted to uppercase
and then compared according to their position in the ANSI character
set. For as <type> comparisons, expressions are compared according
to the current values of sysNumberFormat, sysDateFormat, and
sysTimeFormat as required by the context.

? For details about formats for <type> comparisons, refer to the
“format” topic in online Help.

String operators

Use string operators to concatenate (join) two strings into a longer
one, or to extract substrings from larger strings. For details about
handling text strings, see Chapter 7, “Using variables, values,
and text.”

Operators for string evaluations

Operator Meaning

& Concatenates two expressions with no space between
them. For example:
push (bookPath & "mysysbook.sbk") onto sysBooks

&& Concatenates two expressions with a space between
them. For example:
request "The" && "End" --Displays "The End"

char(s) or One or more contiguous characters of text, including
character(s) letters, numbers, spaces, punctuation marks, tabs, and

other special characters. For example:
if char 1 of userResponse is "Y" then

send exit
end

vFirst3 = chars 1 to 3 of fullName

word or words One or more contiguous sequences of printable charac-
ters separated on each side by a space or a nonprintable
character, or by the beginning or end of the expression
that includes the word. For example:
surname = last word of text of field "FullName"

item or items One or more contiguous values in a comma-separated
list. For example:
xPosition = item 1 of mousePosition of this window
yPosition = item 2 of mousePosition of this window

Operators for string evaluations, continued

Operator Meaning

textline or A string of zero or more characters that ends with a
textlines carriage return/linefeed (CRLF), or two or more strings

separated with CRLFs. For example:
allLines = text of field "listbox1"
step ctr from 1 to textLineCount(allLines)

request textline ctr of allLines
end

Bitwise operators

Bitwise operators allow you to perform binary operations on indi-
vidual bits. They are most useful when you need to read or set
the value of a specific bit in a parameter for a DLL function. For
example, many Windows DLLs return a single value in which each
individual bit designates a particular value. Use numbers to repre-
sent the values for bitwise operators.

Operators for bitwise operations

Operator Meaning

bitAnd Performs a bitwise AND operation on two numbers.
For example:
--Displays 2 (binary 11 AND 10 yields 10) (continued)

request 3 bitAnd 2

bitOr Performs a bitwise inclusive OR operation on two
numbers. For example:
--Displays 3 (binary 01 OR 10 yields 11)
request 1 bitOr 2

bitXOr Performs a bitwise exclusive OR operation on two
numbers. For example:
--Displays 1 (binary 11 XOR 10 yields 01)
request 3 bitXOr 2

bitNot Converts all bits to their opposite values (performs a
one’s complement of a number). For example:
--Displays -2 (binary 00000001 NOT yields 11111110)
request bitNot 1

bitShiftRight Performs a bitwise shift to the right on a number by a
specified number of bits. For example:
--Puts 18 into It (144/(2^3))
get 144 bitShiftRight 3 --(10010000 yields 10010)

bitShiftLeft Performs a bitwise shift to the left on a number by a
specified number of bits. For example:
--Puts 32 into It (4*(2^3))
get 4 bitShiftLeft 3 --(100 becomes 100000)

As an example, the following program calls the DLL function
GetWinFlags( ) and uses bitwise operators to determine the type of
processor in the computer:

to get processorType
local dword winFlags
local retValue
linkDLL "kernel"

DWORD GetWinFlags( )
end linkDLL
winFlags = GetWinFlags( )
conditions

when winFlags bitAnd 2 <> 0
retValue = "286"

when winFlags bitAnd 4 <> 0
retValue = "386"

when winFlags bitAnd 8 <> 0
retValue = "486"

end conditions
return retValue

end processorType

The argument operator

You can use a special OpenScript operator called argument to get
the value of a parameter passed to the currently executing handler.
A parameter can consist of more than one item, so argument n and
item n of argList (where n <= the value of argCount) can return
different values.

--Puts value of the 3rd parameter passed into a field
text of field "Name" = argument 3

Precedence of operators

ToolBook evaluates expressions according to the predefined
operator precedence shown in the following table. Operators of
equivalent precedence are evaluated left-to-right except for the
exponentiation expressions, which are evaluated right-to-left. You
can force ToolBook to change operator precedence by using
parentheses around any expression.

Operator precedence in OpenScript

Order Operators

0 ( )

1 not, argument, – (negation)

2 character, word, item, textline (string operators)

3 bitNot

4 bitShiftLeft, bitShiftRight

5 bitAnd

6 bitOr, bitXOr

7 ^ (exponentiation)

8 *, /, div, mod

9 +, – (subtraction)

10 &, &&

11 is in, is not in, contains

12 <, >, <=, >=, =, <>, is, is not

13 and

14 or

Examples of operator precedence

Example Result

request 1 + 20 / 5 * 40 + 2 163

request (1+20) / (5 * (40 + 2)) .1

request 3^2*2 18

request 3^(2*2)+10 91

request characters 1 to 2 of 1000 + 2000 2010

request characters 1 to 2 of (1000 + 2000) 30

request 1 + 1 & 2 22

request 1 + (1 & 2) 13

Chapter 7

Using variables, values, and text
This chapter describes how you can use individual components
of OpenScript programs, including variables, constants, and literal
values, and discusses how to use text in your scripts.

Using variables . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 118
Assigning data types to variables . . . . . . . . . . . . . . . . . . . 123
Using arrays . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 127
Working with literal values . . . . . . . . . . . . . . . . . . . . . . . . 134
Working with constants. . . . . . . . . . . . . . . . . . . . . . . . . . . 140
Working with character strings . . . . . . . . . . . . . . . . . . . . . 142
Working with lists . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 147
Referring to colors . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 149

Using variables
A variable is a container you define in a script to store data tempo-
rarily. Use a variable any time you are working with data that might
change each time you run the script, as this example shows:

to handle buttonClick
vYearOfBirth = text of field "birthYear" --Stores a property value
vThisYear = "94" --Stores a literal value
vAge = vThisYear - vYearOfBirth --Stores a result
request "This year you will be" && vAge && "years old."

end

Variables in OpenScript are like variables in other programming
languages, with these exceptions:

◆ You can declare variables with or without a data type.

◆ A variable declared without a data type can contain numeric,
logical, string, and other values such as dates and times.

◆ With the exception of arrays, you do not need to declare local
variables before you assign a value to them.

◆ You can define variables as global or local using the local or
system lay words.

You can declare variables to be arrays in OpenScript. For details,
see “Using arrays” later in this chapter.

Rules for naming variables

When you name a variable, the name:

◆ must begin with a letter or underscore.

◆ can contain only alphanumeric characters, underscores, and the
at symbol (@) as a prefix.

◆ cannot be an OpenScript keyword (for example, request or
fillColor).

To distinguish names of variables from names of properties and
OpenScript keywords, use a prefix such as v (variable) or sv (system
variable), or use the reserved prefix @:

local vTotal --Uses "v" to mark local variable

system svVarName --Uses "sv" to mark system variable

@text = text of field "name" --Uses "@" as prefix to distinguish
--variable from a property

Note OpenScript is not case-sensitive, and so the variable names
vStudentName and vstudentname are the same.

Assigning values to variables

Use set, put, or an equal sign (=) to assign a value to a variable;
these commands have the same effect. You can put up to 64 KB
of data into a single variable.

If the value is text and has a space in it, you must enclose the text
in quotation marks. Your script will work more efficiently if you
enclose all text values in quotation marks.

firstPlayer = text of field "Player1"

set PrimaryColors to "red,blue,green"

put "$1,000 cash" into prizeMoney --Contains space,
--requires quotation marks

Using local and system variables

You can use both local and system variables in handlers. A local
variable keeps the value you put into it for the duration of the



120

handler. A system variable retains its value for the duration of the
ToolBook instance. Use a system variable for information that needs
to be available to more than one handler—for example, a user’s
name, the last keystroke entered, or the value of a property before
you change it.

Note You must declare system variables in every handler or
function that uses them.

Creating local variables

You do not have to declare local variables; when you assign the first
value to a variable, ToolBook both declares and initializes the vari-
able. There is a local statement to declare local variables, but it is
used primarily to assign data types to variables.

Tip To share local variable values between handlers, pass them
as parameters.

Creating system variables

To create a system variable, declare it with the system statement.
For example, the following enterPage handler uses a system variable
to determine if the user is navigating to a page for the first time:

to handle enterPage
system svFirstTime --Declares a system variable
if svFirstTime = null then --Null if this is the first time

vWelcomeText = "Welcome to the Tutorial."
text of field "Banner" = vWelcomeText
svFirstTime = false

end if
forward

end enterPage



121

To determine what system variables you have already declared,
get the value of the system property sysSystemVariables, as in this
example:

--Displays all current system variables
to handle buttonClick

if itemCount(sysSystemVariables) > 0
step ctr from 1 to itemCount(sysSystemVariables)

request item ctr of sysSystemVariables
end step

end if
end buttonClick

Tip You can also maintain global values in user properties.
User properties take less memory and are saved with the book.

Using reserved variables

ToolBook maintains certain variables, such as It, argList, and
argCount, into which it puts the results of a statement or information
about the current handler. You can use the following variables in
your scripts. All are treated as local variables within a handler:

◆ A number of statements, including ask, get, pop, request, and set,
put the results of their actions into the variable It. Be sure to copy
the value of It to another container if you need the value for more
than one statement.

◆ The variable argList contains the parameters passed to the current
handler. You cannot change the value of this variable.

◆ The variable argCount contains a count of the number of values
passed as parameters to the current handler. You cannot change
the value of this variable.

For example, in the following handler, the request statement returns
a user response in the variable It. The variable argList returns a

list of four items (two for the location, and one each for the logical
values in isShift and isCtrl), and argCount returns three, because
a buttonClick handler takes three arguments.

to handle buttonClick loc, isShift, isCtrl
request "Do you wish to see arguments?" with "Yes" or "No"
if It is "Yes" then

request "argList =" && argList
request "argCount =" && argCount

end if
end buttonClick

Assigning data types to variables
A data type is a format you assign to a variable when you first
declare it, such as integer, text, real number, and so on. Assigning
data types to variables often makes handlers run faster and use less
memory because data is already in the internal (binary) form that
ToolBook requires, and in its most efficient format. Assigning a data
type to a variable also helps prevent errors because you cannot as-
sign inappropriate data to the variable.

To declare a data type, use the statements local or system with a
data type before referencing the variable:

local LONG i
system INT pagesDisplayed, pagesLeft
local COLOR oldFillColor

If you do not declare a data type for a variable, ToolBook selects
a data type appropriate to the statement in which the variable first
appears. ToolBook can change the data type in subsequent state-
ments as necessary, but this can slow your scripts down. If you do
declare a data type for a variable, ToolBook assigns the appropriate
data type to the variable when you compile the script. When the

handler is running, ToolBook checks that only the correct type of
data is assigned to the variable; if not, an error message displays.

OpenScript data types

Type Description

int 16-bit signed integer, range –32768 to 32767

long 32-bit signed integer, range –2147483648 to 2147483647

real 64-bit floating point number; ToolBook will accept a long
value; all numeric functions and operators that accept real
numbers use this type internally

word 16-bit unsigned integer, range 0 to 65535

dword 32-bit unsigned integer, range 0 to 4294967295

string Character string (Avoid assigning string to data that is more
efficiently represented by a different type.)

logical True or false

point x, y; where x and y are integers (a list of two location values)

color List of three non-negative numbers in HLS units

stack List of comma-separated items

date Date in system format

time Time in system format

page Explicit reference to a page

background Explicit reference to a background

layer Explicit reference to a page or background

graphic Explicit reference to an object on a page or background

field Explicit reference to a field, record field, or combo box

object Explicit reference to a book, page, background, object, or
graphic

book Explicit reference to a book path

Tip Use the function isType( ) to determine whether data matches
a particular data type. For example:

ask "Enter starting date for report:"
if isType(date, It) then

vStartDate = It
else

request "Please enter a date."
end

ToolBook initializes the value of an untyped variable or array to null
when it is declared. ToolBook initializes the value of a typed vari-
able differently depending on the data type, as the following table
shows.

Default values for variables declared with data types

Data type Initial value

dword, int, long, real, word 0

logical false

point 0,0

color 0,0,0

stack, string null

All other types Consider uninitialized

Choosing the appropriate data type

Your application will run more efficiently if you assign a data type
appropriate to how the variable will be used. For example, using the
data type string for a logical value uses more memory than necessary.
Similarly, using the data type string for a counter in a step statement
forces ToolBook to convert the data from a string to a numeric value
with each iteration of the loop. The following table shows opera-
tions that function most efficiently when you use the suggested data
type.

Recommended data types for operations

Operation Data type

Numeric operations real

Text operations word

Logical operations logical

Step loops long

Declaring data types for system variables

You can declare a different data type for system variables in different
handlers. When you assign a value to the variable, ToolBook con-
verts the data to the appropriate type if possible. Because of this
conversion, system variables are slightly less efficient than local
variables. Any reference to the variable assumes that the value is
of the declared type.

Declaring data types for parameters

If you pass a typed variable as a parameter to another handler, you
can declare the data type of parameters in a to handle statement.
This results in greater efficiency and type-checking of data passed
into a handler. For example:

to handle buttonClick point loc, logical isShift, logical isCtrl
… --Further statements here

end

Using arrays
An array is a variable configured to work like a list or table of indi-
vidual elements. For example, if you create a game and want to
store the names of the 10 highest-scoring players, it is easier to have
a single variable called hiScorers with 10 elements in it than it is to
have 10 individual variables called hiScorer1, hiScorer2, hiScorer3,
and so on. Or, if you are writing an application to take concert
bookings for an auditorium, you can use an array to treat the concert
hall as a matrix of rows and seats. Using arrays can help you create
smaller and simpler code because arrays organize information ac-
curately and systematically.

Figure 1  Using arrays to organize information

Use a one-
dimensional array
to represent lists.

Use a two-dimensional
array to represent tables.

You can create one-dimensional arrays and multidimensional arrays
of up to 16 dimensions. You can create arrays with a fixed size or
size them dynamically as you add and remove elements. In general,
use a dynamic array if you do not know in advance how many ele-
ments you require, or if the size of the array will change over time.
You can have both fixed and dynamic arrays in a single handler, but
you cannot mix fixed and dynamic dimensions in the same array. For
example, you cannot use the following line of code:

local logical seat[32][ ]

Handlers often run faster if you use fixed rather than dynamic
arrays. However, because dynamic arrays do not claim a predefined

amount of memory, they sometimes work better under low memory
conditions.

Declaring an array

To declare an array, indicate the dimensions of the array in the local
or system statement. To declare an array that is dynamically sized,
use null values for the number of elements. ToolBook generally runs
faster if you assign a data type to the array.

local hiScorer[10] --Declares array of 10 elements only
local logical seats[32][80] --Declares fixed-size array, 32 rows by 80

--columns; each element is true or false
system users[] --Declares 1-dimensional dynamic array
local playerScores[][] --Declares 2-dimensional dynamic array

The index number of an array begins at one (not zero, as in some
systems).

Setting and getting values in arrays

You can assign values to elements of an array and read their values
as you would any other variable:

auditorium[3][16] = true --Fills one element of the array
if auditorium[5][10] is true

request "That seat is booked."
end

To quickly assign values to all elements at once, use the fill
command, which assigns a single value to the entire array, or
assigns elements one-by-one. If you assign a data type to the array,
be sure that the data type of the fill expression matches the array’s.

For example:

local int barrel[5]
fill barrel with 4 --Assigns the value of 4 to all 5 elements

--of array "barrel"

You can also use in ... order with the fill command to assign individual
units of text—lines, items, words, or characters—to elements in an
array. For example, the fill statement in the following example
assigns each line of a field to an element in an array:

local vHiScorers[]
fill vHiScorers with text of field "hiScore" in [textline] order

Figure 2  Assigning text to an array

T.J.
C.B.
L.S.
P.D.
J.F.
A.N.

Variable: hiScorersField: hiScore

You can use up to four text operators (char, item, textline, and word)
to subdivide text. Each operator you specify breaks the text into
smaller components and assigns them to another dimension of the
array. The number of text operators you use must match the array’s
dimensions, and the array must be large enough to accept all text
components, or ToolBook displays an error message. If the array is
larger than the number of text components, the fill command leaves
the existing contents of the unused elements as they were.

For example, if you are filling a two-dimensional array, you can
assign textlines to one dimension and words to another. The follow-
ing example script results in the array illustrated in Figure 3:

sampleText = "To be or not to be," & CRLF & "that is the question"
local myArray[2][6]
fill myArray with sampleText in [textline][word] order

Figure 3 Assigning text units to an array in textline and word
order

myArray [1] [2] [3]      [4]         [5]      [6]
        [1] To be or       not        to        be,
        [2] that is the     question null     null

You can assign the elements of one array to another using the equal
sign or the set or put command. The arrays must match in dimension
and data type.

local array1[100]
local array2[100]
fill array1 with "a" --Puts "a" into all elements of array1
array2 = array1 --Puts contents of array1 into array2

You cannot assign an array to a container that is not an array.
For example, you cannot assign an array to the Command window.

Sizing arrays

When you declare an array, keep the following in mind:

◆ You can declare up to 65,536 elements in each dimension.

◆ An array can have up to 16 dimensions.

◆ Array size is limited by only the amount of available memory.

Getting an array’s size

To determine an array’s dimensions, use the dimensions function.
Knowing an array’s dimensions is useful if you need to step through
the array element by element. For example:

--Use the value of "dimensions" to set a multiple dimension array
local myArray [5][4][3][2]
vMyDimensions = dimensions(myArray) --vMyDimensions contains
"5,4,3,2"
--Copies all the elements of an array into a text field with
--CRLF between elements
to get arrayToTextlines pArray[]

local vTextlines
get dimensions(pArray) --Puts the size of the array into It
step i from 1 to It

textline i of vTextlines = pArray[i]
end step
return vTextlines

end arrayToTextlines

Passing arrays as parameters

You can pass an array variable as a parameter, as long as the called
handler is written to accept an array for that parameter. To pass an
array as a parameter, specify it the same way you would any variable.
For example:



132

local vAlphabet[26]
get sortArray(vAlphabet)

In the called handler, you indicate an array by including a set of
brackets for each dimension in the handler’s opening statement.
However, when working with dynamic arrays, you do not need to
indicate the size of each dimension, only the number of dimensions,
as in the following example:

--This handler accepts two arrays
to handle modifyArrays array1[], array2[][]

In the following example, a dynamic array is passed as a parameter,
modified, and passed back to the original routine.

--This is the calling handler
to handle buttonClick

local vAlphabet[]
local vInvertedAlphabet[]
--(fill vAlphabet [] here)
vInvertedAlphabet = invertArray ( vAlphabet )
--(further processing here)

end

--The called handler flips the array upside down;
--the array must be declared in the parameter list
to get invertArray fixed vPassedArray[]

--A second array used temporarily in this handler
local vTempArray[]
myDim = dimensions(vPassedArray)
step i from 1 to myDim

--Inverts highest and lowest index numbers

index = abs((i-myDim)-1)
vTempArray[index] = vPassedArray[i]

end step
return vTempArray

end invertArray

Note By default, arrays are passed by value, which means that
ToolBook makes a copy of the array and any changes you make are
not reflected in the calling handler. However, with arrays you can
use the by reference special term to indicate that ToolBook should
pass only a pointer to the original array. This is faster and uses less
memory, but if you change an array in a called handler, it is changed
in the calling handler.

Assigning arrays as user properties

To preserve the values in an array between ToolBook sessions, you
can assign them to user properties just as you can ordinary
variables. You can retrieve an array stored in a user property by
assigning it to an array of the same dimensions. For example:

myUserProp of this page = vMyArray
vMyArray = myUserProp of this page

Working with literal values
A literal value is a value you assign to a variable or use in an
expression. For example:

vStartingValue = 0 --A value assigned to a variable

vGreeting = "Dear" && vName --A literal value in an expression



134

Assigning literal values
to variables with data types

If you assign a literal value to a variable that has been declared
with a data type, the literal value must match the data type for that
variable. If you try to assign a value to a variable that does not
match the data type of that variable, ToolBook will not compile the
script. For example:

local int vHighScore --Declares an integer type
vHighScore = "1000" --Succeeds because "1000" is an integer
vHighScore = "abc" --Will not compile, "abc" is not an integer

Assigning literal values to untyped variables

When you assign a value to a variable that was not declared with
a data type, ToolBook determines the data type for the value
based on how you are using the value. Depending on the context,
ToolBook interprets a value as a string, numeric value, logical
(Boolean) value, time, date, list, color, or point in page units or
pixels. For example:

show group "palmTree" at "100,100" --100,100 is a point in page units

text of field "Total" = sum(100,100) --100,100 is a list

You can take advantage of the ToolBook ability to change the
data type of a value as needed to use the same data in different ways.
For example, you can use numbers as strings, as the following
example shows:

vYear = 1957
request chars 3 to 4 of vYear --Displays "57"
request vYear + 30 --Displays "1987"



135

Specifying numeric values

When you specify numeric values, use only the following characters:

◆ Numeric characters 1, 2, 3, 4, 5, 6, 7, 8, 9, and 0

◆ Decimal sign defined by sysDecimal, such as a period or a comma

◆ Negative and positive operators (– and +)

◆ Scientific notation exponentiation symbol (E or e)

◆ Currency sign defined by sysCurrency, such as $

◆ Percent sign (%)

If you use any other characters, ToolBook treats the value as a
string. For example, ToolBook considers dates in the format
mm/dd/yy strings. If you include decimals when setting the value of
a property that requires integers, ToolBook rounds to the nearest in-
teger. For example, if you set an object’s layer property to 3.75,
ToolBook rounds the number to 4.

Use the round( ) function to round a numeric value to a whole
number or to a number with a specific number of decimal places.
For example:

x = 1.2345 / .333333
request round(x, -4) --Rounds result to 4 decimal places

You can format numbers, dates, and times to display correct char-
acters and to ensure that ToolBook recognizes the value as a num-
ber, date, or time. For details about formatting numbers, dates, and
times, see Chapter 8, “Writing common scripts.”

? Additional information is located in the online Help topics
sysDateformat, sysNumberformat, and sysTimeformat.

Specifying string values

Use string values for text when (for example) you want to assign
captions or text to objects. You can use any combination of charac-
ters for a string value, including alphabetic characters, numbers,
punctuation, and spaces. If you use a string value that contains a
space, you must enclose the value in quotation marks. However,
even if the value does not contain a space, it’s wise to enclose it in
quotation marks. ToolBook will run faster if it does not have to de-
termine whether the literal value is actually a variable. In addition,
you are less likely to cause bugs by accidentally using the name of
an existing variable. For example:

--No space so quotation marks not essential
vMessage = Welcome

--Runs faster with quotation marks
vMessage = "Welcome"

--Requires quotation marks because of spaces
vMessage = "Welcome to the tutorial."

Logical (Boolean) values

Use true and false for logical (Boolean) values. Setting the value of
an object property to true or false is often equivalent to turning an
attribute on or off. For example:

sysUseWindowsColors = true

Specifying page units and pixels

You use page units and pixels to specify values for screen coordinates
for different types of objects. Use page units with graphic objects to
specify an object’s bounds and vertices, the location of an object
relative to the upper-left corner of the page, or the destination or
distance to move an object with the move command:

--Moves button down and right by 1 inch
move button ID 0 by 1440,1440

There are 1440 page units per inch, or about 57 per millimeter.
However, the number of pixels per page unit depends on the display
device. Use the system property sysPageUnitsPerPixel to determine
how many page units there are per pixel for the current display
device. To convert page units for different display devices, use the
displayAspect( ) function.

Use pixels to specify the location or movement of viewers or
palettes. For example:

--Sets maximum size of a viewer to 1/2 size of a VGA monitor
maximumSize of viewer "Index" = 320,240

If you include decimals when specifying page units or pixels,
ToolBook rounds to the nearest integer.

Specifying hexadecimal, octal, and binary values

You can enter hex values using the prefix 0X (zero plus x, not
case-sensitive), as in the following example:

to handle keyChar key
vQuestion = 0x3F
if key = vQuestion then



138

request "Do you wish to display Help?" with "Yes" or "Cancel"
if It is "Yes" then

… --Further statements here
else

forward
end if

else
forward

end if
end keyChar

To format numeric values as hexadecimal, octal, or binary for
display, use the format command with a prefix to indicate the base,
as the following table shows.

Prefixes for specifying numeric values

Base Prefix

hex @h

octal @o

binary @b

decimal @d

The at symbol (@) must appear as the first character of the format

string. For example:

vNumber = 10
format vNumber as "@h00" --Hex conversion
request vNumber --Displays "0A"
vNumber = 10
format vNumber as "@b#" --Binary conversion
request vNumber --Displays "1010"
vNumber = 10
format vNumber as "@o00" --Octal conversion
request vNumber --Displays "12"

? For details on formatting characters, see the “Format” topic (format
command) in online Help.

Working with constants
Constants are words that represent other values, often numbers,
to make the values easier to read and enter. You can use these types
of constants in OpenScript.

Types of constants used in OpenScript

Constant type Examples

Color (HLS or RGB values) red, green, blue

Control characters and punctuation quote, tab, space, CRLF, EOF

Keystrokes keyA, keyUpArrow, keyLeftButton,
keySlash

Mathematical expressions pi

The following examples illustrate constants:

my fillColor = green
--Uses CRLF as constant
text of field "Address" = vAddress & CRLF & vCity
--Uses keyRightButton as constant
if keyState(keyRightButton) is "down" then

… --Further statements here
end

? For a complete list of constants, refer to the “Color constants (table)”
topic in online Help.

For details about the HLS and RGB values used in color constants,
see “Referring to colors” later in this chapter.

About keystroke constants
OpenScript includes keystroke constants that are numeric equivalents
for all keyboard keys and mouse actions (left-button and right-button
clicks). Examples of keystroke constants include:

keyA, keyB keyLeftButton keyTab
keyF1 keyNumpad1 keyUpArrow
keyHome keyNumLock

Use the keystroke constants with the keyState( ) function to determine
whether the user has pressed a specific key or mouse button.
For example:

--This handler displays the unique ID of the object under the
--cursor while the left mouse button is held down
to handle buttonDown

while keyState(keyLeftButton) is down

caption of statusBar = \
objectFromPoint(mousePosition of this window)

end while
end buttonDown

Working with character strings
A character string, or string, is any combination of alphabetic,
numeric, punctuation, and control characters that you assign to
a variable, assign as the property of an object, or use in an
expression. For example:

ask "Enter your complete address, including country"

Note For details about working with lists, see “Working with lists”
later in this chapter.

Using string operators in expressions

To refer to part of a string, use one of the string operators char,
item, textline, or word:

item 1 of newNames --Gets one item out of a list in a variable

char 2 of word 3 of textline 4 of text of field "Note" of \
page 27 of book "home.tbk"

You can identify part of a string, or a substring, using an absolute
number (as shown above) or using a keyword to indicate its position:
first, last, middle (or mid), second, third, fourth, up to tenth. For
example:

third character of vAlphabet --Gets "C" from a local variable

last item of vNames --Gets last item in a local variable

For details about string operators, see Chapter 6, “Creating an
OpenScript statement.”

Tip Whenever possible, use the char operator to refer to part of
a string because it executes much faster than word or textline.

Using a range

To identify a range within a string, use the keyword to. For example:

vIntro = words 1 to 3 of text of field "Preamble of Constitution"

This statement sets the variable vIntro to We, the people, including
the commas.

Converting between strings and other data types

When you use string operators with data in untyped variables,
ToolBook automatically converts the data to a string. For example:

x = (1000*2)
vFirstChar = first character of x --Returns "2"

You can also use strings as numbers in expressions. In the following
expression, ToolBook converts the text of a field into a number:

request (word 1 of text of field "Tax Rate" of page 7) * vAmount



143

You can produce a list by joining strings with a comma and then as-
signing the string to a variable:

vAlphabet = "A,B,C,D,E"

You can use the list commands (pop, push, clear, get, put, and set) to
manipulate lists.

Note You can convert between strings, numbers, and other data
types only if the data is stored in variables that were not declared
using data types.

Inserting or replacing text in strings

You can use string specifiers to replace text or to insert text into a
string, as the table on the next page shows. The new value does not
have to be the same size as the old value.

String specifiers for inserting or replacing strings

String specifier Description

into Replaces a string

before Inserts text in front of another string

after Inserts text behind another string

For example:

--To replace the first word of a string with a phrase
word 1 of textline 2 of text of field "Habitat" = "Large shady trees"

--To insert the word "shady" after the first word, and move any
--other words to the right
put "shady" after word 1 of textline 2 of text of field "Habitat"

Working with textlines

The string operator textline returns a line of text, but does not return
the carriage return/line feed (CRLF) that delimits the line. If you
insert a new textline before or after an existing textline, use the
constant CRLF to include a carriage return/linefeed with the new
textline. For example:

--To start a new line and insert the specified string
put CRLF & "average annual rainfall: 40 inches" after \

textline 3 of text of field "Habitat"

If you need to manipulate the text of a field, put the text into a
variable first. Extracting data from a variable is faster than extract-
ing it from properties of an object. For example:

vTextOfField = text of field "Introduction"
step ctr from 1 to wordCount(vTextOfField)

request word ctr of vTextOfField
end

Selecting text in a field or list box

To select a string within the text of a field, use the select command:

select words 2 to 5 of text of field "Instructions"

When text is selected, you can act on it by referring to the
selectedText system property:

select word 1 of text of field "Instructions"
strokeColor of selectedText = red --Sets text to red

The system property selectedTextState returns a list of information
(such as the offset within a field) about the currently selected text.

To determine what lines have been selected in a list box, use the
field’s selectedTextlines property. You can also use selectedTextlines
to select particular lines from within a handler to set, for example,
default values in a list box. The property selectedTextlines contains
a list of one or more line numbers representing the selected lines.
For example:

--This handler is in a button (not in the script of the list box)
--and displays the text of lines that have been clicked in a list box
to handle buttonClick

vTextOfField = text of field "listbox1"
vSelectionList = selectedTextlines of field "listbox1"
step ctr from 1 to itemCount(vSelectionList)

vNextLineNo = item ctr of vSelectionList
vNextLine = textline vNextLineNo of vTextOfField
request vNextLine

end step
end buttonClick

Functions used with strings

You can use the following OpenScript functions to refer to or act on
specific parts of a string:

ansiToChar( ) lowercase( ) uppercase( )

charCount( ) offset( ) wordCount( )
charToAnsi( ) textFromPoint( )
itemCount( ) textlineCount( )

For example, the following statement puts a list of all but the first
three finalists into a variable:

vConsolation = items 4 to itemCount(vFinalists) of vFinalists

When manipulating text in a field, improve processing speed by
putting the text into a variable and performing the required
operations on the variable. String functions are especially useful for
such operations. For example:

--To find out if any character in a field is not a number
--vTxt is a local variable
vTxt = text of field "RefValues" of this background
step i from 1 to charCount(vTxt)

if character i of vTxt is not in "0123456789"
return false

end if
end step

Working with lists
A list, or stack, is a series of items separated by commas. Lists are
a good way to maintain data that consists of a series, such as
names, especially if the number of items changes often. You can
also use lists to store a series of items temporarily without having
to create separate variables for each item.

ToolBook has special commands and functions that work with lists,
and often requires parameters to be passed as lists (such as page
units or color values). You can refer to each element of a list using
the keyword item:

vNames = "Amelia,Molly,Isabel,Roxanne"
get item 2 of vNames --Puts the name "Molly" into It

Use the commands push and pop to treat a list as a first-in-last-out
stack. Push adds an item (including the comma delimiter) to the
front of a container; pop removes it. For example:

local vGrades --Initializes to null
push "A" onto vGrades
push "B" onto vGrades --vGrades now contains "B,A"
--Puts "B" into the variable It (vGrades now contains only "A")
pop vGrades
pop vGrades into vLastItem --Puts "A" into vLastItem

You can use push to position items at a specific place in the stack:

push "dog" onto item 5 of vAnimals
push "cat" after item 5 of vAnimals

When you refer to items in a list, you can put a working copy of
the list into a variable and pop items from the variable until it is null
rather than use a step control structure to iterate through the origi-
nal list. For example:

system svSysPositions
local vObjWinner, vMyList
--Puts value of system variable into a local variable
push svSysPositions onto vMyList
step i from 1 to itemCount(vMyList) --Checks each item in the list

pop vMyList
if i <> It

break
end if

end step

Referring to colors
A color in ToolBook is specified by a list of three numbers, repre-
senting either RGB (red, green, and blue) or HLS (hue, lightness, and
saturation) units. You can set colors directly using these values or
use predefined color constants for some colors such as red, blue,
black, and white. You can also set a property for objects that causes
them to use the Windows system colors.

You can set colors in OpenScript using either RGB or HLS units. To
set a color using RGB units, set the rgbFill or rgbStroke properties of
an object; to use HLS units, set the fillColor and strokeColor property.
There is no difference in the result, so use the color system that you
are most comfortable with. You can alternate RGB and HLS units in
the same application, and even use RBG units one time and HLS
units another time for the same object.

Specify RGB values as follows:

◆ Red values range from 0 to 255, with 0 representing 0 percent red
and 255 representing 100 percent red. Pure red in RGB values is
represented by 255,0,0.

◆ Green values range from 0 to 255, with 0 representing 0 percent
green and 255 representing 100 percent green. Pure green in RGB
values is represented by 0,255,0.

◆ Blue values range from 0 to 255, with 0 representing 0 percent

blue and 255 representing 100 percent blue. Pure blue in RGB
values is represented by 0,0,255.

White in RGB units is created by combining 100 percent red, green,
and blue and is represented by 255,255,255. Black is created by
combining 0 percent red, green, and blue and is represented by
0,0,0.

Specify HLS values as follows:

◆ Hue values range from 0 to 360, corresponding to the angle on
a color wheel. For pure colors, 0° is red, 60° is yellow, 120° is
green, 180° is cyan, 240° is blue, and 300° is magenta. Intermediate
angles represent intermediate colors. 0° and 360° are equivalent.

◆ Lightness values range from 0 to 100, with 0 representing
0 percent lightness (black) and 100 representing 100 percent
lightness (white).

◆ Saturation values range from 0 to 100, with 0 representing 0 per-
cent saturation (gray) and 100 representing 100 percent saturation
(pure color).

--Display in sequence all colors available on the current display
--(in HLS units)
to handle buttonClick

step vHue from 0 to 360
my fillColor = vHue,50,100

end step
end buttonClick

Note You can get the RGB and HLS values of a color by using the
Color dialog box.

Referring to color constants

You can use 10 ToolBook color constants to refer to solid colors—
black, blue, cyan, green, magenta, red, yellow, white, gray, and
lightGray. Each constant is equivalent to the three numeric
values that constitute either the HLS or RGB values for that color.
ToolBook assigns the correct values depending on whether you
are assigning the color to a property that uses HLS or RGB values.
For example:

my fillColor = green --Sets to 120,50,100 (HLS)
rgbFill of rectangle "patch" = green --Sets to 0,255,0 (RGB)
my strokeColor = 0,50,100 --Red (HLS)

Using Windows system colors

To color an object using the Windows-defined colors for that object
type, set the object’s useWindowsColors property to true. This over-
rides the current settings for strokeColor and fillColor, but preserves
these values. If you later set the value of useWindowsColors to false,
ToolBook reverts to the values in strokeColor and fillColor.

